Complex Variables with MAPLE

By

Gilberto E. Urroz, Ph.D., P.E.

Distributed by

infoClearinghouse.com

16 Complex variables

Complex numbers involve the unit imaginary number, i, which results from taking the square roots of negative numbers. Incorporating the imaginary number into variables produces complex variables. In this chapter we present operations with complex numbers and introduce the calculus of complex variables.

Introduction to complex numbers

A *complex number z* is a number written as

$$z = x + iy,$$

where x and y are real numbers, and I is the *imaginary unit* defined by $I^2 = -1$.

The complex number $x+iy$ has a *real part*,

$$x = Re(z),$$

and an *imaginary part*,

$$y = Im(z).$$

We can think of a complex number as a point $P(x,y)$ in the x-y plane, with the x-axis referred to as the *real axis*, and the y-axis referred to as the *imaginary axis*. Thus, a complex number represented in the form $x+iy$ is said to be in its *Cartesian representation*.

A complex number can also be represented in polar coordinates (*polar representation*) as

$$z = re^{i\theta} = r\cdot\cos\theta + I\ r\cdot\sin\theta$$

where

$$r = |z| = (x^2+y^2)^{1/2}$$

is the *magnitude* of the complex number z, and

$$\theta = Arg(z) = \arctan(y/x)$$

is the *argument* of the complex number z.

The relationship between the Cartesian and polar representation of complex numbers is given by the **Euler formula**:

$$re^{i\theta} = \cos\theta + I\sin\theta$$

The *complex conjugate* of a complex number $z = x + iy = re^{i\theta}$, , is

$$\overline{z} = x - iy = re^{-I\theta}.$$

The complex conjugate of z can be thought of as the reflection of z about the real (x-) axis. Similarly, the *negative* of z,

$$-z = -x-iy = -re^{i\theta},$$

can be thought of as the reflection of z about the origin.

Examples of basic complex number operations in Maple

The unit imaginary number, i = √(-1), is represented in Maple by the symbol I. A sequence of the first 10 powers of I can obtained by using:

```
> seq(I^j,j=1..10);
```

I, -1, -I, 1, I, -1, -I, 1, I, -1

A complex number, say $z = 3+5i$, is written as:> z:= 3+5*I;

The functions *Re* and *Im* can be used to obtain the real and imaginary parts, respectively, of a complex number, for example:

```
> Re(z);Im(z);
```

3
5

The magnitude and argument of the complex number *z* are obtained as:

```
> abs(z);argument(z);
```

sqrt(34)
arctan(5/3)

The polar representation of a complex number is accomplished by using the function *polar*. This function needs to be read with *readlib* before it can be used. The following example illustrates its application:

```
> readlib(polar);polar(z);
```

polar(sqrt(34), arctan(5/3))

Alternatively, you can use the option polar with the function *convert*:

```
> convert(z,polar);
```

polar(sqrt(34), arctan(5/3))

The complex conjugate of a complex number is obtained by using the function *conjugate*:

```
> conjugate(z);
```

3 - 5 I

The negative is simply obtained by adding a minus sign to the number, i.e.,

```
> -z;
```

-3 - 5 I

The functions presented here can be applied to a complex expression such as $z = a+bi$, e.g.,

```
> z1:=a+b*I;abs(z1);argument(z1);convert(z1,polar);conjugate(z1);-z1;
```

z1 := a + I b
| a + I b |
argument(a + I b)
polar(| a + I b |, argument(a + I b))
$\overline{a + I\,b}$
-a - I b

Notice, however, that the results produced by applying functions such as *abs, argument*, and *conjugate*, are symbolic results representing the required functions as would be written in paper when applied to complex numbers. To force an expanded result for the functions use *evalc*, i.e.,

```
> evalc(abs(z1));evalc(argument(z1));evalc(conjugate(z1));
```

$$\sqrt{a^2+b^2}$$

$$\arctan(b, a)$$

$$a - I\,b$$

To plot a complex number using Maple, use the function *complexplot* from the *plots* package. The argument of *complexplot* is a list of complex numbers in their Cartesian form, e.g.:

```
> with(plots):
> complexplot([z,conjugate(z),-z],x=-6..6, y = -6..6, style=point, symbol = circle, color = blue);
```



Complex number calculations

Complex numbers can be added, subtracted, multiplied, and divided. The rules for these operations are shown below:

Let $z = x + i\cdot y = r\cdot e^{i\cdot\theta}$, $z_1 = x_1 + i\cdot y_1 = r_1\cdot e^{i\cdot\theta_1}$, and $z_2 = x_2 + i\cdot y_2 = r_2\cdot e^{i\cdot\theta_2}$,

be complex numbers. In these definitions the numbers x, y, x_1, x_2, y_1, and y_2 are real numbers.

Addition: $z_1 + z_2 = (x_1 + x_2) + i\cdot (y_1 + y_2)$

Subtraction: $z_1 - z_2 = (x_1 - x_2) + i\cdot (y_1 - y_2)$

Multiplication: $z_1\cdot z_2 = (x_1\cdot x_2 - y_1\cdot y_2) + i\cdot(x_1\cdot y_2 + x_2\cdot y_1) = r_1\cdot r_2\cdot e^{i\cdot(\theta_1+\theta_2)}$.

Multiplication of a number by its conjugate results in the square of the number's magnitude, i.e.:

$$z\cdot \bar{z} = (x + i\cdot y)\cdot(x - i\cdot y) = x^2 + y^2 = r^2 = |z|^2$$

Division:

$$\frac{z_1}{z_2} = \left(\frac{z_1}{z_2}\right)\cdot\left(\frac{\bar{z}_2}{\bar{z}_2}\right) = \frac{z_1\cdot\bar{z}_2}{|z_2|^2} = \frac{x_1\cdot x_2 + y_1\cdot y_2}{x_2^2 + y_2^2} + i\,\frac{y_1\cdot x_2 - x_1\cdot y_2}{x_2^2 + y_2^2} = \frac{r_1}{r_2}\cdot e^{i\cdot(\theta_1-\theta_2)}.$$

Powers: $z^n = (r\cdot e^{\,i\cdot\theta})^n = r^n\cdot e^{\,i\cdot n\theta}$

Roots: because the argument θ of a complex number z has a periodicity of 2π, we can write

$$z = r\cdot e^{\,i\cdot(\theta+2k\pi)}, \quad \text{for } k = 0,\ 1,\ 2,\ \ldots$$

There are n n-th roots of z calculated as

$$\sqrt[n]{z} = z^{1/n} = r^{1/n}\cdot e^{i\cdot\frac{(\theta+2k\pi)}{n}}, \quad k = 0,1,2,\cdots(n-1).$$

Examples of operations with complex numbers

The following are operations with complex numbers using Maple:

```
> z1:=-5+2*I;z2:=3+4*I; #Defining two complex numbers
```

z1 := -5 + 2 I
z2 := 3 + 4 I

Addition, subtraction, multiplication, and division:

```
> z1+z2;z1-z2;z1*z2;z1/z2;
```

-2 + 6 I
-8 - 2 I
-23 - 14 I

$$-\frac{7}{25} + \frac{26}{25}\,I$$

The following sequence provides the first 5 integer powers of z1:

```
> seq(z1^j,j=1..5);
```

-5 + 2 I, 21 - 20 I, -65 + 142 I, 41 - 840 I, 1475 + 4282 I

This is an attempt to find the cubic root of z1. Function *evalc* is used to expand the result in complex form. The function *evalf* is used to provide the floating-point version of the complex number:

> z1^(1/3);evalc(%);evalf(%);

$$(-5+2\,I)^{\left(\frac{1}{3}\right)}$$

$$29^{\left(\frac{1}{6}\right)}\cos\left(-\frac{1}{3}\arctan\left(\frac{2}{5}\right)+\frac{1}{3}\pi\right)+I\,29^{\left(\frac{1}{6}\right)}\sin\left(-\frac{1}{3}\arctan\left(\frac{2}{5}\right)+\frac{1}{3}\pi\right)$$

1.061378105 + 1.394916967 I

The last three commands can be put together in the following manner:

> evalf(evalc(z1^(1/3)));

1.061378105 + 1.394916967 I

The previous expression provides only the principal cubic root of z1. To find all the roots of z^n, write an equation of the form $z^n = z1$ and solve it using the function solve, for example:

> *z:='z':eqz:=z^3=z1;solve(eqz,z);*

$$eqz := z^3 = -5+2\,I$$

$$(-5+2\,I)^{\left(\frac{1}{3}\right)},\ -\frac{1}{2}(-5+2\,I)^{\left(\frac{1}{3}\right)}-\frac{1}{2}I\sqrt{3}\,(-5+2\,I)^{\left(\frac{1}{3}\right)},\ -\frac{1}{2}(-5+2\,I)^{\left(\frac{1}{3}\right)}+\frac{1}{2}I\sqrt{3}\,(-5+2\,I)^{\left(\frac{1}{3}\right)}$$

The results provided consist of the three roots of z1. To express them in their final Cartesian form using floating point numbers use the function *evalf*:

> evalf(%);

1.061378105 + 1.394916967 I, .6773444775 - 1.616638886 I, -1.738722583 + .2217219185 I

Functions of a complex variable

We defined a complex variable z as $z = x + iy$, where x and y are real variables, and $i = (-1)^{1/2}$. We can also define another complex variable

$$w = F(z) = \Phi + i\Psi,$$

where, in general,

$$\Phi = \Phi(x,y), \text{ and } \Psi = \Psi(x,y),$$

are two real functions of (x, y). These real functions can also be given in terms of the polar coordinates (r,θ) if we use the polar representation for z, i.e.,

$$z = r\cdot e^{i\theta} = r\,(\cos\theta + i\cdot\sin\theta).$$

In such case,

$$\Phi = \Phi(r,\theta), \text{ and } \Psi = \Psi(r,\theta).$$

Recall that the coordinate transformations between Cartesian and polar coordinates are:

$$r = (x^2 + y^2)^{1/2}, \qquad \tan\theta = y/x,$$
$$x = r\cos\theta, \qquad y = r\sin\theta$$

The complex variable *w* is also known as a *complex function*. Another name for a complex function is "*mapping*." Thus, we say *F(z)* is a mapping of *z*. In geometric terms, this means that any figure in the x-y plane gets "mapped" onto a different figure on the Φ–Ψ plane by the complex function F(z).

As an example, take the function

$$w = F(z) = \ln(z) = \ln(r \cdot e^{i\theta}) = \ln(r) + i\theta.$$

We can identify the functions

$$\Phi = \Phi(r,\theta) = \ln(r), \text{ and } \Psi = \Psi(r,\theta) = \theta,$$

as the real and imaginary components, respectively, of the function *ln(z)*. Using the transformations indicated above we can also write,

$$\Phi = \Phi(x,y) = \ln[(x^2+y^2)^{1/2}] = (1/2)\ln(x^2+y^2), \text{ and } \Psi = \Psi(x,y) = \tan^{-1}(y/x).$$

Expanding *ln(z)* using Maple

The following commands define the function w = ln(z) and attempt to obtain its real and imaginary components:

```
> z:='z':z:=x+I*y;w:=ln(z);phi:=Re(w);psi:=Im(w);
```

$$z := x + I\,y$$
$$w := \ln(x + I\,y)$$
$$\phi := \ln(|\,x + I\,y\,|)$$
$$\psi := \operatorname{argument}(x + I\,y)$$

The expressions for $\phi(x,y)$ and $\psi(x,y)$ obtained above present the appropriate expressions before evaluation as complex expressions in their Cartesian form. Use of evalc permits defining $\phi(x,y)$ and $\psi(x,y)$) in their Cartesian form. The next two commands are used to define $\phi(x,y)$ and $\psi(x,y)$ as functions:

```
> phi:=(x,y)->evalc(Re(w));psi:=(x,y)->evalc(Im(w));
```

$$\phi := (x,\ y) \to \operatorname{evalc}(\operatorname{Re}(w))$$
$$\psi := (x,\ y) \to \operatorname{evalc}(\operatorname{Im}(w))$$

To see the expression for the functions we can use:

```
> phi(x,y);psi(x,y);
```

$$\ln(\sqrt{x^2 + y^2})$$

$$\arctan(y, x)$$

Plots of the functions $\phi(x,y)$ and $\psi(x,y)$ follow:

```
> plot3d(phi(x,y),x=-1..1,y=-1..1,axes=boxed);
```



> plot3d(psi(x,y),x=-1..1,y=-1..1,axes=boxed);



Derivative of a complex function

The derivative of the complex variable f(z), to be referred to as f'(z) = df/dz, is, by definition,

$$f'(z) = \frac{df}{dz} = \lim_{\Delta z \to 0} \frac{f(z+\Delta z) - f(z)}{\Delta z}.$$

The definition of a complex derivative requires us to evaluate the function f(z) at a point P(x,y) corresponding to z = x + iy, and at point Q(x+Δx, y+Δy), as illustrated in the figure below.



The figure also illustrates the fact that to get from point z to point z+Δz in the complex x-y plane you can follow a multitude of paths. In general, the value of the derivative will depend on the path we follow to define Δz. Because we want the derivative df/dz to be uniquely defined, we need to find some criteria such that, regardless of the path selected to define z, the value of df/dz remains the same.

In general, we will write Δz = Δx+iΔy. Let's calculate the derivative df/dz utilizing paths for Δz along the x-axis alone, i.e, Δz = Δx, and along the y-axis alone, i.e., Δz = iΔy. Thus, for Δz = Δx, we can write

$$\frac{df}{dz} = \lim_{\Delta x \to 0} \frac{[\Phi(x+\Delta x, y) + i\Psi(x+\Delta x, y)] - [\Phi(x,y) + i\Psi(x,y)]}{\Delta x}$$

$$\frac{df}{dz} = \lim_{\Delta x \to 0}\left(\frac{[\Phi(x+\Delta x, y) - \Phi(x,y)]}{\Delta x} + i\frac{[\Psi(x+\Delta x, y) - \Psi(x,y)]}{\Delta x} \right)$$

i.e.,

$$\frac{df}{dz} = \frac{\partial \Phi}{\partial x} + i\frac{\partial \Psi}{\partial x}.$$

You can prove, by expressing the derivative in terms of Δz = iΔy, that

$$\frac{df}{dz} = \frac{\partial \Psi}{\partial y} - i\frac{\partial \Phi}{\partial y}.$$

In order for the last two expressions for df/dz to be the same, we require that

$$\frac{\partial \Phi}{\partial x} = \frac{\partial \Psi}{\partial y}, \qquad \frac{\partial \Phi}{\partial y} = -\frac{\partial \Psi}{\partial x}.$$

These two equations are known as the *Cauchy-Riemann differentiability cond*itions for complex functions (or, simply, the Cauchy-Riemann conditions). Thus, if the functions $\Phi(x,y) = Re[f(z)]$ and $\Psi(x,y) = Re[f(z)]$, satisfy the Cauchy-Riemann conditions, the derivative $f'(z)=df/dz$ is uniquely defined. In such case, the function f(z) is said to be an *analytical complex function,* and the functions Φ(x,y) and Ψ(x,y) are said to be *harmonic functions*.

More importantly, if a complex function f(z) is analytical, the rules used for univariate derivatives can be applied to f(z). For example, iwe indicated earlier that the function

$$w = f(z) = ln\ (z) = ln\ (r \cdot e^{i\theta}) = ln(r) + i\theta.$$

can be written in terms of (x,y) as

$$\Phi = \Phi(x,y) = ln[(x^2+y^2)^{1/2}] = (1/2)\ ln(x^2+y^2), \text{ and } \Psi = \Psi(x,y) = tan^{-1}(y/x).$$

Using Maple, let's check if the functions Φ(x,y) and Ψ(x,y) satisfy the Cauchy-Riemann conditions:

```
> simplify(diff(phi(x,y),x));simplify(diff(psi(x,y),y));#Try it in Maple
```

i.e., $\partial\Phi/\partial x = \partial\Psi/\partial y$, checks out ok.

```
> simplify(diff(phi(x,y),y));simplify(diff(psi(x,y),x));#Try it is Maple
```

also, $\partial\Phi/\partial y = - \partial\Psi/\partial x$, checks out ok.

Alternatively, you can use the function *Is* as follows:

```
is(diff(phi(x,y),x)=diff(psi(x,y),y));
                              true

> is(diff(phi(x,y),y)=-diff(psi(x,y),x));
                              true
```

The function f(z) = ln(z) is, therefore, analytical, and its derivative can be calculated by using:

$$\frac{df}{dz} = \frac{d}{dz}(\ln z) = \frac{1}{z}.$$

In Maple, we use the following commands to prove the previous equation:

```
> z:='z':f:=z->ln(z);diff(f(z),z);

                          f := ln
                            1/z
```

You can define the derivative as a Maple function, *fp(z),* as follows:

```
> fp:=z -> diff(f(z),z);fp(z);
```

fp := z -> diff(f(z), z)
1/z

To express the result in terms of *z* = *x*+*iy* use:

```
> subs(z=x+I*y,fp(z));
```

$$\frac{1}{x+I\,y}$$

To obtain a complex Cartesian representation for *fp(z)* we can use:

```
> simplify(evalc(subs(z=x+I*y,fp(z))));
```

$$\frac{x-I\,y}{x^2+y^2}$$

Earlier on we obtained the result:

$$\frac{df}{dz}=\frac{\partial\Phi}{\partial x}+i\frac{\partial\Psi}{\partial x}.$$

This expression can be verified using Maple as follows:

```
> simplify(evalc((diff(phi(x,y),x)+I*diff(psi(x,y),x))));
```

$$\frac{x-I\,y}{x^2+y^2}$$

Also, f'(z) can be obtained from:

$$\frac{df}{dz}=\frac{\partial\Psi}{\partial y}-i\frac{\partial\Phi}{\partial y}.$$

Which in Maple can be evaluated as:

```
> simplify(evalc((diff(psi(x,y),y)-I*diff(phi(x,y),y))));
```

$$\frac{x-I\,y}{x^2+y^2}$$

Note: Most of the functions that we commonly use with real variables, e.g., *exp, ln, sin, cos, tan, asin, acos, atan*, hyperbolic functions, polynomials, *1/x*, square root, etc., are analytical functions when used with the complex variable *z* = *x* +*iy*. Thus, the rules of derivatives for these functions are the same as in real variables, e.g., *d(sin(z))/dz = cos(z), d(z^2+z)/dz = 2z+1*, etc.

Multivariate calculus and complex function applications in potential flow

The concepts of partial derivatives and derivative of a complex variable have practical applications in the analysis of potential or ideal flow in two-dimensions. Ideal flow refers to the flow of a fluid that has no viscosity (inviscid fluid), while potential flow stands for a flow whose velocity components are obtained as partial derivatives of a function $\phi(x,y)$, called the flow potential function. Ideal flow and potential flow are synonyms.

Continuity equation

The equation of continuity is the mathematical expression of the law of conservation of mass for fluid flow. Considering an inviscid, incompressible (constant density) fluid flow in two dimensions. The equation of continuity for these flows is given by

$$\partial u/\partial x + \partial v/\partial y = 0,$$

Where $u = u(x,y)$, $v = v(x,y)$ are the x- and y-components of flow velocity in the plane.

Stream function

Let us define a function $\psi(x,y)$ such that the velocity components u and v are

$$u = \partial\psi/\partial y \text{ and } v = -\partial\psi/\partial x.$$

If we replace this function into the continuity equation we have

$$\partial(\partial\psi/\partial y)/\partial x + \partial(-\partial\psi/\partial x)/\partial y = 0,$$

or

$$\partial^2\psi/\partial y\partial x - \partial^2\psi/\partial x\partial y = 0,$$

which is satisfied by any continuous function $\psi(x,y)$. The function $\psi(x,y)$ is known as the *stream function* of the flow.

Curves defined by $\psi(x,y) = constant$ are known as the *streamlines* of the flow. The velocity vector

$$\mathbf{q}(x,y) = u(x,y)\cdot\mathbf{i}+v(x,y)\cdot\mathbf{j}$$

at any point (x,y) on a streamline is tangent to the streamline.

The total differential for the stream function along a streamline $\psi(x,y)$ = constant is

$$d\psi = (\partial\psi/\partial x)\cdot dx + (\partial\psi/\partial y)\cdot dy = 0.$$

Therefore, the slope of the streamline at a point (x,y) is given by

$$m_\psi = dy/dx = -(\partial\psi/\partial x)/(\partial\psi/\partial y) = -(-v)/u=v/u.$$

Potential flow

A flow whose velocity components are obtained from

$$u = \partial\phi/\partial x, \text{ and } v = \partial\phi/\partial y,$$

where $\phi(x,y)$ is a scalar (i.e., non-vector) function, is referred to as a *potential flow*, and the function $\phi(x,y)$ is known as the *velocity potential*.

If we replace the definitions of u and v into the continuity equation, what results is the following partial differential equation known as Laplace's equation:

$$\partial^2\phi/\partial x^2 + \partial^2\phi/\partial y^2 = 0.$$

Curves defined by $\phi(x,y) = constant$ are known as the *iso-potential or equipotential lines* of the flow. The total differential for the velocity potential along a equipotential line $\phi(x,y)$ = constant is

$$d\phi = (\partial\phi/\partial x)\cdot dx + (\partial\phi/\partial y)\cdot dy = 0.$$

Therefore, the slope of the equipotential line at a point (x,y) is given by

$$m_\phi = dy/dx = -(\partial\phi/\partial x)/(\partial\phi/\partial y) = -u/v.$$

The flow net

The fact that the slope of a streamline is given by $m_\psi=v/u$, and that of an equipotential line is given by $m_\phi=-u/v$, indicates that at the point of intersection of any two of these lines the lines are normal to each other. This follows from the fact that

$$m_\psi\cdot m_\phi=(v/u)\cdot(-u/v) = -1,$$

which is the condition for two straight lines to be perpendicular to each other. In this case the straight lines of interest are the tangential lines to the streamline and to the equipotential line at the point of intersection.

A picture of a collection of equipotential lines and streamlines is known as a *flow net*.

Irrotational flow

When a fluid particle is subjected to motion it undergoes not only translation, but also suffers elongation (normal strains), shear strains, and rotation. In two dimensions, you can prove that the magnitude of the angular velocity of a fluid particle in a flow is given by

$$\omega = (\partial v/\partial x - \partial u/\partial y).$$

A fluid flow where the fluid particles undergo no rotation is called an *irrotational flow*. For such a flow we have $\omega = 0$, or

$$\partial v/\partial x - \partial u/\partial y = 0.$$

Replacing the velocity components in terms of the stream function ($u = \partial\psi/\partial y$, $v = -\partial\psi/\partial x$) reveals the fact that $\psi(x,y)$ also satisfies Laplace's equation, i.e.,

$$\partial^2\psi/\partial x^2 + \partial^2\psi/\partial y^2 = 0.$$

Example 1 – Verify that a fluid flow whose velocity components are given by $u = x/(x^2+y^2)$, $v = y/(x^2+y^2)$, satisfy the continuity equation and the condition of irrotationality. Also, determine expressions for the potential function $\phi(x,y)$ and the stream function $\psi(x,y)$.

First, we define the continuity equation as:

```
> restart:ContEq:=diff(u(x,y),x)+diff(v(x,y),y) = 0;
```

$$ContEq := \left(\frac{\partial}{\partial x}\mathrm{u}(x,y)\right)+\left(\frac{\partial}{\partial y}\mathrm{v}(x,y)\right)=0$$

Next, define the functions:

```
> restart:u:=(x,y)->x/(x^2+y^2);v:=(x,y)->y/(x^2+y^2);
```

To verify if the velocity components, $u(x,y)$, and $v(x,y)$, satisfy the continuity equation simply type:

```
> simplify(ContEq);
```

$$0 = 0$$

The result is indeed zero, thus proving that $u(x,y)$ and $v(x,y)$ satisfy the continuity equation.

The condition of irrotationality is given by the expression,

$$\partial v/\partial x - \partial u/\partial y = 0.$$

which, in Maple can be set up as:

```
> IrrotCond:=diff(v(x,y),x)-diff(u(x,y),y);
```

$$IrrotCond := 0$$

To find the velocity potential we start from $u(x,y) = \partial\phi/\partial x = x/(x^2+y^2)$, which we can integrate with respect to x to obtain

$$\phi(x,y) = \int u(x,y)\cdot dx + F(y) = \int \frac{x}{x^2+y^2}\cdot dx + F(y)$$

By using

```
> phi:=int(u(x,y),x)+F(y);
```

$$\phi := \frac{1}{2}\ln(x^2+y^2)+\mathrm{F}(y)$$

Next, we use the fact that $v(x,y) = \partial\phi/\partial y$.

```
> eq1:=diff(phi,y) = v(x,y);eq2:=diff(F(y),y) = solve(eq1,diff(F(y),y));
```

$$eq1 := \frac{y}{x^2+y^2}+\left(\frac{\partial}{\partial y}\mathrm{F}(y)\right)=\frac{y}{x^2+y^2}$$

$$eq2 := \frac{\partial}{\partial y}\mathrm{F}(y)=0$$

The latter equation, *eq2*, implies $F(y) = K$, where K is a constant. Thus, we can write

$$\phi(x,y) = \tfrac{1}{2}\cdot ln(x^2+y^2) + K.$$

To find the stream function we start from $u(x,y) = \partial\psi/\partial y = x/(x^2+y^2)$, which we can integrate with respect to x to obtain

$$\psi(x,y) = \int u(x,y)\cdot dy + G(x) = \int \frac{x}{x^2+y^2}\cdot dy + G(x).$$

By using

```
> psi:=int(u(x,y),y)+G(y);
```

$$\psi := \arctan\left(\frac{y}{x}\right) + G(y)$$

Next, we use the fact that $v(x,y) = -\partial\psi/\partial x$.

```
> eq3:=-diff(psi,x)=v(x,y);eq4:=diff(G(x),x) = solve(eq3,diff(G(x),x));
```

$$eq3 := \frac{y}{x^2\left(1+\frac{y^2}{x^2}\right)} - \left(\frac{\partial}{\partial x}G(x)\right) = \frac{y}{x^2+y^2}$$

$$eq4 := \frac{\partial}{\partial x}G(x) = 0$$

This implies $G(y) = C$, where C is a constant. Thus, we can write

$$\psi(x,y) = tan^{-1}(y/x) + C.$$

To visualize the flow net for this case, first we need to select values of the constants K and C. We can stipulate that the point $(x,y) = (0,0)$ belongs to the streamline $\psi = 0$, to make $C = 0$. Similarly, we can force point $(0,0)$ into the equipotential line $\phi = 0$, to make $K = 0$. The result for the velocity potential and stream function for these conditions are

$$\phi(x,y) = \tfrac{1}{2}\cdot ln(x^2+y^2), \text{ and } \psi(x,y) = tan^{-1}(y/x).$$

The flow net can be drawn by using contour plots as follows:

```
> with(plots):
> plot1:=contourplot(phi(x,y),x=-5..5,y=-5..5, scaling=constrained):
> plot2:=contourplot(psi(x,y),x=-5..5,y=-5..5, scaling = constrained):
> display(plot1,plot2,axes=boxed);
```



Complex potential and complex velocity

The complex function

$$F(z) = \phi(x,y) + i \cdot \psi(x,y)$$

is referred to as the *complex potential* of the flow.

Recalling that the derivative of this complex function can be written as

$$dF/dz = \partial\phi/\partial x + i \cdot \partial\psi/\partial x = \partial\psi/\partial y - i \cdot \partial\phi/\partial y,$$

and from the definition of the velocity components u and v, it follows that dF/dz, referred to as the *complex velocity*, w(z), contains the velocity components in its real and imaginary parts. The complex velocity is written as

$$w(z) = dF/dz = u - i \cdot v.$$

Thus, $u = Re(w)$, and $v = -Im(w)$.

Elementary two-dimensional potential flows

Because the equations governing the potential flow phenomena are linear equations (Laplace's equation), you can obtain the complex potential of a flow by adding the complex potentials of elementary flows. In this section we present the complex potentials of some elementary flows such as *uniform flow, source and sink, vortex*, and *doublet*. The last three are known as *singularity flows* since the velocities go to infinity at the location of the singularity generating the flow. A doublet is simply the combination of a source and a sink of the same strength that are infinitesimally close to each other. The strength of a singularity is a measure related to the flow discharge into a source or out of a sink, or to the angular velocity of a vortex.

The following are the complex potentials for these elementary flows:

Uniform flow with streamlines parallel to the x-axis:	$F(z) = U \cdot z$
Source ($m>0$) or sink ($m<0$) of strength m located at $(0,0)$:	$F(z) = m \cdot \ln z$
Vortex of strength G ($G>0$, counterclockwise) at $(0,0)$:	$F(z) = i \cdot G \cdot \ln z$
Doublet of strength μ ($\mu > 0$, if sink is located to the left of source):	$F(z) = \mu/z$

We can obtain the velocity potential and stream function of any of these flows by using

$$\phi = Re[F(z)] \text{ and } \psi = Re[F(z)].$$

Example 1 -- To find the real and imaginary part of F(z) = ln z, a source flow with strength m = 1, we use the result obtained earlier,i.e.,

$$\phi = \phi(x,y) = ln[(x^2+y^2)^{1/2}] = (1/2)\ ln(x^2+y^2),\ \text{and}\ \psi = \psi(x,y) = tan^{-1}(y/x)$$

Example 2 – Find the velocity potential and stream function for the complex potential *F(z) = U·z+m/z*.

Use:

```
> restart:F:=z->U*z+m/z;
```

F := z -> U z + m/z

```
> phi:=evalc(Re(subs(z=x+I*y,F(z))));
```

$$\phi := U\,x + \frac{m\,x}{x^2+y^2}$$

```
> psi:=evalc(Im(subs(z=x+I*y,F(z))));
```

$$\psi := U\,y - \frac{m\,y}{x^2+y^2}$$

Example 3 – For the complex potential used in Example 2 obtain expressions for the components of velocity u and v.

Use:

```
> w:=diff(F(z),z);
```

$$w := U - \frac{m}{z^2}$$

The velocity component *u(x,y)* is the real part of *w(z)*:

```
> u:=evalc(Re(subs(z=x+I*y,w)));
```

$$u := U - \frac{m\,(x^2-y^2)}{(x^2-y^2)^2 + 4\,x^2\,y^2}$$

The velocity component *v(x,y)* is the imaginary part of *w(z)*:

```
> v:=evalc(Im(subs(z=x+I*y,w)));
```

$$v := 2\,\frac{m\,x\,y}{(x^2-y^2)^2 + 4\,x^2\,y^2}$$

We can simplify the expressions for u and v by checking that the denominator can be written as:

```
> factor(expand((x^2-y^2)^2+4*x^2*y^2));
```

$$(x^2+y^2)^2$$

Thus, the velocity components simplify to:

$$u = U - \frac{m(x^2-y^2)}{(x^2+y^2)^2}, \quad v = \frac{2mxy}{(x^2+y^2)^2}.$$

Plotting the complex potential

To plot the real and imaginary parts of the complex potential you can identify the real and imaginary components $\phi(x,y)$ and $\psi(x,y)$ of the complex potential and use contour plots to plot the flow net as illustrated in the example below:

For example, to produce a Gridmap plot for the function $F(z) = 1/z$, use the following:

```
> with(plots):
> plot1:=contourplot(phi,x=-1..1,y=-1..1, contours=20, color = blue):
> plot2:=contourplot(psi,x=-1..1,y=-1..1, contours = 20, color = red):
> display({plot1,plot2},axes = boxed, scaling=constrained);
```



Complex potential for combinations of elementary flows

When we add complex potentials of elementary flows we can obtain the picture of more complicated flows. Some of those combined flows are presented here.

<u>*Example 1*</u> – Find the velocity components for the combination of two sources both of strength $m = 1$, one located at $x = -1$ (source s_1), the other at $x = +1$ (source s_2). The complex potentials corresponding to the two sources are

$$F_1(z) = \ln(z+1),\ F_2(z) = \ln(z-1).$$

The combined complex potential is

$$F(z) = F_1(z) + F_2(z) = \ln(z+1) + \ln(z-1).$$

Use:

```
> restart:F:=z->ln (z +1) + ln(z-1);
```

$$F := z \to \ln(z+1) + \ln(z-1)$$

```
> phi:=evalc(Re(subs(z=x+I*y,F(z))));psi:=evalc(Im(subs(z=x+I*y,F(z))));
```

$$\phi := \ln(\sqrt{x^2+2x+1+y^2}) + \ln(\sqrt{x^2-2x+1+y^2})$$

$$\psi := \arctan(y, x+1) + \arctan(y, x-1)$$

```
> with(plots):
> plot1:=contourplot(phi,x=-2..2,y=-2..2, contours=10, color = blue):
> plot2:=contourplot(psi,x=-2..2,y=-2..2, contours = 20, color = red):
> display({plot1,plot2},axes = boxed, scaling=constrained);
```



Example 2 – Find the velocity components for the combination of combination of a uniform flow (U=1) and a negative doublet of strength $\mu = 1$ (i.e., the source is located to the left of the sink in the doublet on the x-axis). The complex potentials corresponding to the two sources are

$$F_1(z) = z, \; F_2(z) = 1/z.$$

The combined complex potential is

$$F(z) = F_1(z) + F_2(z) = z + 1/z.$$

Use:

```
> restart:F:=z->z+1/z;
> phi:=evalc(Re(subs(z=x+I*y,F(z))));
> psi:=evalc(Im(subs(z=x+I*y,F(z))));
```



The graph shows the equipotential lines and streamlines of the combination of a uniform flow (U=1) and a negative doublet of strength $\mu = 1$ (i.e., the source is located to the left of the sink in the doublet on the x-axis). The graph on the right-hand side has been modified by shadowing the region within a circle of radius 1, which happens to constitute a closed streamline. For all practical purposes you can replace the flow within the closed streamline with a solid body, in this case a cylinder. Thus, the combination of this uniform flow and negative doublet produces the flow net corresponding to a uniform flow U = 1 past a cylinder of radius 1.

Plotting the velocity field

The function *gradplot* within the *plots* package, applied to the real component of the complex potential, $\phi(x,y) = Re(F(z))$, produces plots of the velocity vector $\mathbf{q} = \nabla\phi(x,y)$. For example, for the complex potential $F(z) = z^2$ produces the following velocity vector plot:

```
restart:F:=z->z^2;
```

$$F := z \rightarrow z^2$$

```
> phi:=evalc(Re(subs(z=x+I*y,F(z))));
```

$$\phi := x^2 - y^2$$

```
> with(plots):gradplot(phi,x=-2..2,y=-2..2);
```



This plot shows the arrows representing velocity vectors as thin lines. Use of the option *arrows = thick* in the *gradplot* function call produces a plot where the velocity vectors are more clearly defined.

```
> gradplot(phi,x=-2..2,y=-2..2,arrows=thick);
```



The option color in the gradplot function call allows the use of color in the arrows. The value of the color in the example shown here are determined by the function $\phi(x,y)$ itself.

```
> gradplot(phi,x=-2..2,y=-2..2,arrows=thick,color=phi);
```



Taylor series for complex functions

Taylor series were introduced earlier in Chapter for functions of real variables. It was indicated there that the Taylor series expansion of the function *f(x)* about the point $x = x_0$ can be written as

$$f(x) = \sum_{n=0}^{\infty} \frac{f^{(n)}(x_0)}{n!} \cdot (x - x_0)^n.$$

In Maple, the function *series(f(x),x=x0)* is used to obtain a Taylor polynomial, *P(x),* and the order of the remainder, *R(x)*, from the expansion:

$$f(x) = P_k(x) + R_k(x) = \sum_{n=0}^{k} \frac{f^{(n)}(x_0)}{n!} \cdot (x - x_0)^n + R(x).$$

Recall that the order of the remainder is determined by the value of the environmental variable *Order*, whose default value is 6.

For complex variables the same command *series* can be used to expand a complex function about a value $z=z_0$. For example, to expand the function *f(z) = ln(z)* about the point $z_0=I$, use:

```
> restart:f:=z->ln(z):z0:=I:ff:= series(f(z),z=z0);
```

$$ff := \frac{1}{2} I \pi - I(z-I) + \frac{1}{2}(z-I)^2 + \frac{1}{3} I (z-I)^3 - \frac{1}{4}(z-I)^4 - \frac{1}{5} I (z-I)^5 + O((z-I)^6)$$

To convert the series to a polynomial use the function convert as follows:

```
> F:= convert(ff, polynom);
```

$$F := \frac{1}{2} I \pi - I(z-I) + \frac{1}{2}(z-I)^2 + \frac{1}{3} I (z-I)^3 - \frac{1}{4}(z-I)^4 - \frac{1}{5} I (z-I)^5$$

The *radius of convergence* of the series is given by the radius of the largest circle in the complex plane about the point $z=z_0$ where the series is analytic. One way to estimate this radius of convergence is by using the ratio test. The radius of convergence is defined as

$$\rho = \lim_{n \to \infty} \left| \frac{a_n}{a_{n+1}} \right|,$$

where a_n is the *n*-th non-zero element in the Taylor series expansion.

The coefficients of the Taylor series expansion can be obtained by using the function *coeftayl*. This function needs to be invoked with the function readlib before it can be used. For example, for the function F defined above we can use:

```
> readlib(coeftayl);
```

The coefficient of the Taylor series expansion of function F about $z = I$ for $n = 2$ is calculated from:

```
> coeftayl(F,z=I,2);
```

$$\frac{1}{2}$$

To determine the radius of convergence we produce the sequence of values $|a_n/a_{n+1}|$ for $n = 1,2, \ldots$, to see if it converges to a constant value. This can be implemented with Maple by using the following commands. The value of the Maple environmental variable *Order* is set to 20 for this example. First, we produce the Taylor polynomial as we did earlier and store it into variable F:

```
> restart:Order:=20:f:=z->ln(z):z0:=I:ff:=series(f(z),z=z0):F:= convert(ff, polynom):
```

Next, we load the function *coeftayl* and define the function *a(n)* as the *n*-th coefficient of the Taylor polynomial F:

```
> readlib(coeftayl):a:=n->coeftayl(F,z=I,n);
```

a := n -> coeftayl(F, z = I, n)

A sequence of values of *a(n)* is shown next:

```
> seq(a(n),n=1..20);
```

$$-I, \frac{1}{2}, \frac{1}{3}I, \frac{-1}{4}, -\frac{1}{5}I, \frac{1}{6}, \frac{1}{7}I, \frac{-1}{8}, -\frac{1}{9}I, \frac{1}{10}, \frac{1}{11}I, \frac{-1}{12}, -\frac{1}{13}I, \frac{1}{14}, \frac{1}{15}I, \frac{-1}{16}, -\frac{1}{17}I, \frac{1}{18}, \frac{1}{19}I, 0$$

The following is a sequence of values $|a_n/a_{n+1}|$:

```
> seq(evalf(abs(a(n)/a(n+1))),n=1..18);
```

2., 1.500000000, 1.333333333, 1.250000000, 1.200000000, 1.166666667, 1.142857143, 1.125000000, 1.111111111, 1.100000000, 1.090909091, 1.083333333, 1.076923077, 1.071428571,1.066666667, 1.062500000, 1.058823529, 1.055555556

The results of the ratio test indicate that as n grows, the ratio $|a_n/a_{n+1}|$ tends to the value *1.0*, thus, the radius of convergence for the Taylor series $f(z) = ln(z)$ about $z = I$ is $\rho = 1.0$.

Singular points, poles, and Laurent series for complex functions

A singular point of a complex function *f(z)* is a value $z = z_0$ where the function fails to be analytic. The point $z = z_0$ is referred to as an *isolated singularity* of *f(z)*. For example, for the function *f(z) = ln z* the point *z=0* is a singular point.

If a complex function *f(z)* can be written in the form

$$\mathrm{f}(z) = \frac{\phi(z)}{(z - z_0)^n} \;,$$

where $\phi(z)$ is analytical in a region that contains the point $z = z_0$, and *n* is a positive integer, then *f(z)* is said to have a pole of order *n* at $z = z_0$. Thus, the function

$$\mathrm{f}(z) = \frac{z}{z^2 + 16} = \frac{z}{z} - 4\,\mathrm{i}(z + 4\,i) \;,$$

has two poles of order 1, one at *z = 4i* and one at *z=-4i*.

If a complex function *f(z)* has a pole of order *n* at $z = z_0$ but it is otherwise analytic within the domain $|z - z_0| \le R$, then *f(z)* can be written as the following Laurent series

$$\mathrm{f}(z) = \frac{a_{-n}}{(z - z_0)^n} + \frac{a_{-n+1}}{(z - z_0)^{(n-1)}} + \ldots + \frac{a_{-1}}{z - z_0} + a_0 + a_1(z - z_0) + a_2\,(z - z_0)^2 + \ldots$$

The part of the series corresponding to $a_0 + a_1(z - z_0) + a_2\,(z - z_0)^2 + \ldots$ is referred to as the *analytic part* of the series, while the remaining part, $\mathrm{f}(z) = \frac{a_{-n}}{z - z_0^{\;n}} + \frac{a_{-n+1}}{(z - z_0)^{(n-1)}} + \ldots + \frac{a_{-1}}{z - z_0}$, is referred to as the *principal part* of the series. The Laurant series can be written in compact form as

$$\mathrm{f}(z) = \sum_{k=-\infty}^{\infty} a_k\,(z - z_0)^k.$$

A function that is analyitic in a region bounded by two concentric circles and centered at $z = z_0$, i.e., $R_1 < |z - z_0| < R_2$, can always be expanded into a Laurent series. The Maple function *series*, used earlier to produce Taylor series, also produces Laurent series if the expansion of the series is taken about a singularity of the function. For example, the function, $\mathrm{f}(z) = \frac{\mathbf{e}^z}{(z-1)^2}$, can be expanded about the point $z = 1$ by using:

```
> restart:f:=z->exp(z)/(z-1)^2;z0:=1;
```

$$f := z \to \frac{e^{z}}{(z-1)^2}$$

$$z0 := 1$$

```
> series(f(z),z=z0);
```

$$e\,(z-1)^{-2} + e\,(z-1)^{-1} + \frac{1}{2}e + \frac{1}{6}e\,(z-1) + \frac{1}{24}e\,(z-1)^2 + \frac{1}{120}e\,(z-1)^3 + O((z-1)^4)$$

If a higher order for the remainder is required, the environmental variable *Order* can be modified, for example:

```
> Order:=8;series(f(z),z=z0);
```

$$Order := 8$$

$$e\,(z-1)^{-2} + e\,(z-1)^{-1} + \frac{1}{2}e + \frac{1}{6}e\,(z-1) + \frac{1}{24}e\,(z-1)^2 + \frac{1}{120}e\,(z-1)^3 + \frac{1}{720}e\,(z-1)^4 + \frac{1}{5040}e\,(z-1)^5 + O((z-1)^6)$$

Notice in both versions of the Laurent series that the expansion in negative powers of $(z-1)$ has a finite number of terms, with the largest negative power (*n=2*) corresponding to the order of the pole $z=1$ of the function. The series expansion is valid for any complex number $z \neq 1$, therefore, we say that the series converges for any value of $z \neq 1$.

Laurent series expansions about a singularity in a complex function can be used to define the type of singularity under consideration. For example, if the principal part of the series expansion (i.e., the part corresponding to negative powers of the expansion) has finite number of terms reaching up to terms of power -*n*, then we have a pole of order *n.* The example presented above corresponds to a pole of order 2.

As a second example of Laurent series expansion use the function $f(z) = \frac{\sin(z)}{z-\pi}$. The function has a singularity at point $z=\pi$, therefore, we attempt an expansion of the function about that point by using the function *series*:

```
> restart:Order:=10;f:=z->sin(z)/(z-Pi);series(f(z),z=Pi);
```

$$Order := 10$$

$$f := z \to \frac{\sin(z)}{z-\pi}$$

$$-1 + \frac{1}{6}(z-\pi)^2 - \frac{1}{120}(z-\pi)^4 + \frac{1}{5040}(z-\pi)^6 - \frac{1}{362880}(z-\pi)^8 + O((z-\pi)^9)$$

The expansion contains no negative powers of the term $(z-\pi)$, therefore, the point $z=\pi$ is referred to as a *removable singularity*. The expansion will converge for all values of z.

If the principal part of a Laurent series expansion has an infinity number of terms, the singularity is said to be an *essential singularity.* For example, the Laurent series expansion for the function $f(z) = e^{\left(\frac{1}{z}\right)}$ can be attempted by using:

```
> restart:f:=z->exp(1/z);series(f(z),z=0);
```

$$f := z \to e^{\left(\frac{1}{z}\right)}$$

```
Error, (in series/exp) unable to compute series
```

A direct application of the function *series* produces an inconclusive result. Instead, we are going to use the expansion of the function $f(y) = \mathbf{e}^y$ and then replace $y = \frac{1}{z}$, i.e.,

```
> restart:f:=y->exp(y);fs:=series(f(y),y=0);
```

$$f := \exp$$

$$fs := 1 + y + \frac{1}{2}y^2 + \frac{1}{6}y^3 + \frac{1}{24}y^4 + \frac{1}{120}y^5 + O(y^6)$$

```
> subs(y=1/z,fs);
```

$$1 + \frac{1}{z} + \frac{1}{2}\frac{1}{z^2} + \frac{1}{6}\frac{1}{z^3} + \frac{1}{24}\frac{1}{z^4} + \frac{1}{120}\frac{1}{z^5} + O\left(\frac{1}{z^6}\right)$$

This is a Laurent series expansion with an infinity number of terms in the principal part, therefore, the pole $z = 0$ is an *essential singularity* or a *pole of infinite order*.

Integrals of complex functions

If $f(z) = \phi(x, y) + i\,\psi(x, y)$ is a complex analytic function in a given region of the plane, we define the integral of *f(z)* along a path C from point $z_1 = x_1 + i\,y_1$ to point $z_2 = x_2 + i\,y_2$, as:

$$\int_C f(z)dz = \int_{z_1}^{z_2} f(z)dz = \int_{(x_1,y_1)}^{(x_2,y_2)} [\phi(x,y) + i\psi(x,y)]\cdot(dx + idy)$$

$$= \int_{(x_1,y_1)}^{(x_2,y_2)} (\phi dx - \psi dy) + i\int_{(x_1,y_1)}^{(x_2,y_2)} (\psi dx + \phi dy)$$

For example, to calculate the integral of the complex function $f(z) = \mathbf{e}^z$, along the path $y = x + 1$ between points $z_1 = -2 + i$ and $z_2 = 1 + 2\,i$, use:

```
> restart:f:=z->exp(z);phi:=evalc(Re(subs(z=x+I*y,f(z))));psi:=evalc(Im(subs(z=x+I*y,f(z))));
```

$$f := \exp$$

$$\phi := \mathbf{e}^x \cos(y)$$

$$\psi := \mathbf{e}^x \sin(y)$$

```
> phi_s:=subs(y=x+1,phi);psi_s:=subs(y=x+1,psi);#substituting y = x + 1
```

$$phi\_s := \mathbf{e}^x \cos(x + 1)$$

$$psi\_s := \mathbf{e}^x \sin(x + 1)$$

with $dy = dx$, the integrals defined above become now $\int_{x_1}^{x_2} \phi - \psi\,dx + i\int_{x_1}^{x_2} \psi + \phi\,dx$, i.e.,

```
> 'int(phi_s-psi_s,x=-2..1)+I*int(psi_s+phi_s,x=-2..1)';%;evalf(%);
```

$$\int_{-2}^{1} phi\_s - psi\_s\,dx + I\int_{-2}^{1} psi\_s + phi\_s\,dx$$

$$\mathbf{e}\cos(2) - \mathbf{e}^{(-2)}\cos(1) + I\,(\mathbf{e}\sin(2) + \mathbf{e}^{(-2)}\sin(1))$$

$$-1.204326349 + 2.585607386\, I$$

Because the function $f(z) = e^z$ is analytic throughout the entire x-y plane we can attempt the following integration:

```
> 'int(f(z),z=-2+I..1+2*I)';%;evalf(%);
```

$$\int_{-2+I}^{1+2I} f(z)\, dz$$

$$e^{(1+2I)} - e^{(-2+I)}$$

$$-1.204326350 + 2.357845958\, I$$

We get the same result as in the previous approach. This latter result suggests that it is not necessary to evaluate the integral in terms of the variables (x,y), instead we simply evaluate the integral of the complex function $f(z)$ between the limits of integration z_1 and z_2. Such a result is possible only if the integral is independent of the path. We can prove that if the function $f(z)$ is analytic, then the resulting integrals

$$\int_{(x_1,y_1)}^{(x_2,y_2)} (\phi dx - \psi dy) + i \int_{(x_1,y_1)}^{(x_2,y_2)} (\psi dx + \phi dy)$$

are independent of the path. The conditions for independence of the path for these two integrals are $\frac{\partial}{\partial y}\phi = -\left(\frac{\partial}{\partial x}\psi\right)$ and $\frac{\partial}{\partial x}\psi = \frac{\partial}{\partial y}\phi$, respectively, which are the Cauchy-Riemann conditions necessary for analycity:

$$\frac{\partial \Phi}{\partial x} = \frac{\partial \Psi}{\partial y}, \qquad \frac{\partial \Phi}{\partial y} = -\frac{\partial \Psi}{\partial x}.$$

Thus, integrals of any analytic function $f(z)$ can be evaluated as a simple univariate integral treating the function z as a real variable and evaluating the complex integration limits, $z_1 = x_1 + i\, y_1$, and $z_2 = x_2 + i\, y_2$, once the anti-derivative of $f(z)$ has been found.

As an example, to calculate the integral $\int_{-2+i}^{i-1} \sin(z)\, dz$ use:

```
> 'int(sin(z),z=-2+I..I-1)';%;evalf(%);
```

$$\int_{-2+I}^{-1+I} \sin(z)\, dz$$

$$-\cos(1)\cosh(1) - I\sin(1)\sinh(1) + \cos(2)\cosh(1) + I\sin(2)\sinh(1)$$

$$-1.475878150 + .0797097159\, I$$

Integrals on closed curves

Let C be a simple closed curve, and let *f(z)* be analytic within the region bounded by C as well as along C itself. The *Cauchy theorem* for complex functions states that

$$\oint_C f(z) = 0$$

This is reasonable considering that the integral of an analytic function is independent of the path, therefore, integrating about a close curve with the upper limit of integration being equal to the lower limit of integration results in the integral being zero.

If the function to be integrated along a simple closed curve C is such that a singularity exists within the region bounded by C then the integral is not equal to zero, instead, the following *Cauchy's integral formulas* apply:

Let $\mathrm{f}(z)$ be analytic within and on a simple closed curve C and let $z = z_0$ be any point interior to C, then

$$\oint_C \frac{f(z)}{z - z_0} = 2\pi i f(z_0)$$

and

$$\oint_C \frac{f(z)}{(z - z_0)^n} = \frac{2\pi i f^{[n]}(z_0)}{n!}$$

where the curve is transversed in the positive (counterclockwise) direction for the integration, and $f^{[n]}$(x) = $\frac{d^n f}{dx^n}$

.

For example, to evaluate the integral of the function $\frac{z^2}{z - \pi}$ along the closed curve given by the circle $|z - \pi| < 2$, we use:

```
> restart:f:=z->z^2;integral:=2*Pi*I*f(Pi);
```

$$f := z \to z^2$$

$$integral := 2\, I\, \pi^3$$

To calculate the integral of the function $\frac{\mathbf{e}^z}{(z-1)^3}$ along the closed curve given by the circle $|z - 1| < 1$, use:

```
> restart:f:=z->exp(z);integral:=2*Pi*I*subs(z=1,diff(f(z),z$3))/3!;
```

$$f := \exp$$

$$integral := \frac{1}{3}\, I\, \pi\, \mathbf{e}$$

Residues and residue theorem for integrals

If a complex function *f(z)* can be expanded into a Laurent series of the form

$$\mathrm{f}(z) = \frac{a_{-n}}{(z-z_0)^n} + \frac{a_{-n+1}}{(z-z_0)^{(n-1)}} + \ldots + \frac{a_{-1}}{z-z_0} + a_0 + a_1(z-z_0) + a_2(z-z_0)^2 + \ldots$$

the coefficient a_{-1} in the series is referred to as the residue of *f(z)* at the pole $z = z_0$. For a pole of order *n* the residue can be calculated using the formula

$$a_{-1} = \lim_{z \to z_0} \frac{1}{(n-1)!} \frac{d^{n-1}}{dz^{n-1}} [(z-z_0)^n f(z)]$$

For a pole of order 1, the formula simplifies to

$$a_{-1} = \lim_{z \to z_0} [(z-z_0) f(z)]$$

For example, to determine the residue of the function $\mathrm{f}(z) = \frac{\sin(z)}{(z-2)^3}$ for $z_0 = 2$ we use:

```
> restart:f:=z->sin(z)/(z-2)^3; n:=3; z0:=2;
```

$$f := z \to \frac{\sin(z)}{(z-2)^3}$$

$$n := 3$$

$$z0 := 2$$

```
> 'diff((z-2)^n,z$(n-1))';%;
```

$$\mathrm{diff}((z-2)^n, z\,\$\,(n-1))$$

$$6z - 12$$

```
> 'limit((1/(n-1)!)*diff((z-2)^n*f(z),`$`(z,(n-1))),z=2)';%;
```

$$\lim_{z \to 2} \frac{\mathrm{diff}((z-2)^n\,\mathrm{f}(z), z\,\$\,(n-1))}{(n-1)!}$$

$$-\frac{1}{2}\sin(2)$$

This residue can be obtained from the Laurent series expansion:

```
> ### WARNING: `residue` might conflict with Maple's meaning of that name
  series(f(z),z=2);residue:=coeff(%,(z-2)^(-1));
```

$$\sin(2)(z-2)^{-3} + \cos(2)(z-2)^{-2} - \frac{1}{2}\sin(2)(z-2)^{-1} - \frac{1}{6}\cos(2) + \frac{1}{24}\sin(2)(z-2) + \frac{1}{120}\cos(2)(z-2)^2 + \mathrm{O}((z-2)^3)$$

$$residue := -\frac{1}{2}\sin(2)$$

The residue theorem states that if *f(z)* is analytic within and on a closed curve C except for a finite number of poles $z_1, z_2, \ldots, z_n$ within the region R limited by C such that the corresponding residues are $a_{1,-1}, a_{2,-1}, \ldots, a_{n,-1}$, then the integral of *f(z)* on the closed curve C is given by

$$\oint_C f(z)dz = 2\pi i(a_{1,-1} + a_{2,-1} + \cdots + a_{n,-1})$$

Notice that the Cauchy integral formulae presented earlier are special cases of this residue theorem.

For example, consider the function $\mathrm{f}(z) = \frac{z^2}{z^3 + 4z - 2z^2 - 8}$. To find the poles of the function we factor the denominator of the function, i.e.,

```
> factor(z^3+4*z-2*z^2-8);
```

$$(z-2)(z^2+4)$$

The function *factor*, as used here, only factors in the real domain. Using the qualifier *complex* will factor the expression down to complex roots, i.e.,

```
> factor(z^3+4*z-2*z^2-8,complex);
```

$$(z+2.\,I)\,(z-2.\,I)\,(z-2.000000000)$$

The poles are, therefore, the values $z = -2\,i$, $z = 2\,i$, and $z = 2$. These values can be found also by using the function *solve* on the denominator of the function, i.e.,

```
> solve(z^3+4*z-2*z^2-8);
```

$$2,\, 2\,I,\, -2\,I$$

Since the three poles found are of order 1, the residues can be found by using the formula

$$a_{-1} = \lim_{z \to z_0} [(z - z_0)\, f(z)]$$

In Maple, we use:

```
> f:=z->z^2/(z^3+4*z-2*z^2-8);
```

$$f := z \to \frac{z^2}{z^3 + 4\,z - 2\,z^2 - 8}$$

```
> a[1,-1]:=limit((z-2)*f(z), z=2);a[2,-1]:=limit((z-2*I)*f(z),z=2*I);a[3,-1]:=limit((z+2*I)*f(z),z=-2*I);
```

$$a_{1,\,-1} := \frac{1}{2}$$

$$a_{2,\,-1} := \frac{1}{4} - \frac{1}{4}\,I$$

$$a_{3,\,-1} := \frac{1}{4} + \frac{1}{4}\,I$$

Thus, the integral of *f(z)* along a simple closed curve C that contains the three poles is obtained as:

```
> '2*Pi*I*sum(a[k,-1],k=1..3)';integral:=2*Pi*I*sum(a[k,-1],k=1..3);
```

$$2\,I\,\pi \left(\sum_{k=1}^{3} a_{k,\,-1} \right)$$

$$integral := 2\,I\,\pi$$

Calculating residues using the Maple function residue

Maple includes the function *residue*, which should be loaded with *readlib*, to calculate residues of complex functions. For example, the residue of the function $f(z) = \frac{\sin(z)}{(z-2)^3}$ for $z_0 = 2$ can be calculated as follows:

```
> ### WARNING: persistent store makes one-argument readlib obsolete
restart:readlib(residue):f:=z->sin(z)/(z-2)^3;z0:=2;a[-1]:=residue(f(z),z=z0);
```

$$f := z \rightarrow \frac{\sin(z)}{(z-2)^3}$$

$$z0 := 2$$

$$a_{-1} := -\frac{1}{2}\sin(2)$$

Evaluation of definite integrals using the residue theorem

The results from the residue theorem can be used to solve definite integrals in real variables by a suitable choice of the integration curve C. For example, the integral $\int_{-\infty}^{\infty} f(x)\, dx$ can be calculated by evaluating the integral $\oint_C f(z)dz$ along the closed curve C illustrated in the figure below:



The closed curve C includes both the diameter -R < x < R along the x-axis, referrred to as C_1, and the semicircle $x^2 + y^2 = R^2$, referred to as C_2. The integration is carried out, as indicated, in a counterclockwise direction. After the integration of the complex function *f(z)* is performed, the integral is calculated by letting $R \rightarrow \infty$.

Before presenting an application example, however, we need to prove the following assertion: If $|f(z)| \leq \frac{M}{R^k}$, where $F(z)$ is a complex function, $z = R\, \mathbf{e}^{(i\,\theta)}$, M and k are constants and $1 < k$, then $\lim_{R\rightarrow\infty} \int_{C_2} f(z)dz = 0$. The curve C_2 is the semi-circle travelled in the counterclockwise direction. The proof requires us to use the fact that, for a function $f(z)$, the following relationship holds:

$$\left|\int_{C_2} f(z)dz\right| \leq \int_{C_2} |f(z)| \cdot |dz| \leq \frac{M}{R^k} \cdot \pi \cdot R = \frac{\pi M}{R^{k-1}}$$

Thus, if the constants M and k exist, and $1 < k$, as $R \rightarrow \infty$ the upper bound of the expression found above will

become zero.
The integral about the closed curve $C = C_1 + C_2$ can be split as follows:

$$\oint_C f(z)dz = \int_{C_1} f(x)dx + \int_{C_2} f(z)dz$$

Along the diameter C_1, y = 0, and $z = x + i\,y = x$, thus, the integral of $f(z)$ gets replaced by that of $f(x)$. We can also write

$$\int_{-R}^{R} f(x)dx = \oint_C f(z)dz - \int_{C_2} f(z)dz$$

Thus, as $R \to \infty$, we have

$$\lim_{R\to\infty} \int_{-R}^{R} f(x)dx = \lim_{R\to\infty} \oint_C f(z)dz - \lim_{R\to\infty} \int_{C_2} f(z)dz$$

If the result $\lim_{R\to\infty} \int_{C_2} f(z)dz = 0$ holds, then the original integral, $\int_{-\infty}^{\infty} f(x)\,dx$ gets calculated as:

$$\int_{-\infty}^{\infty} f(x)dx = \lim_{R\to\infty} \int_{-R}^{R} f(x)dx = \lim_{R\to\infty} \oint_C f(z)dz.$$

For example, consider the integral $\int_0^{\infty} \frac{1}{1+x^4}\,dz$. To use the procedure outlined above we need to verify first if the condition $\left|f(z)\right| \le \frac{M}{R^k}$ for $f(z) = \frac{1}{1+z^4}$. We can do that by replacing $z = R$ so that $\left|f(R)\right| = \frac{1}{1+R^4} \le \frac{1}{R^4}$. Thus, for $M = 1$ and $k = 4$, the function $f(z) = \frac{1}{1+z^4}$ satisfies the condition required to use the procedure outline above for calculating the integral under consideration.
The procedure using Maple is as follows:

```
> restart:F:=z->1/(1+z^4);
```

$$F := z \to \frac{1}{1+z^4}$$

The poles of the function are obtained from:

```
> z0:=[solve(1+z^4,z)];
```

$$z0 := \left[\frac{1}{2}\sqrt{2} + \frac{1}{2}I\sqrt{2}, -\frac{1}{2}\sqrt{2} - \frac{1}{2}I\sqrt{2}, \frac{1}{2}\sqrt{2} - \frac{1}{2}I\sqrt{2}, -\frac{1}{2}\sqrt{2} + \frac{1}{2}I\sqrt{2}\right]$$

A plot of the region of integration and the poles follows. Because plots require a specific value of R, we arbitrarily use R = 2 for the plot:

```
> with(plots):fig1:=implicitplot(x^2+y^2=2^2,x=-2..2,y=0..2,color=red,scaling=constrained):#semi-circle
```

```
> fig2:=pointplot([[-2,0],[2,0]],style = line,color = red):#diameter
```

```
> fig3:=complexplot(z0,x=-3..3,y=-2..3,style=point, symbol=circle,color=blue):#poles
```

```
> display({fig1,fig2,fig3});#combined plot - diameter shadowed by axes
```



There are only two poles, $z=-\frac{1\sqrt{2}}{2}+\frac{1\,I\sqrt{2}}{2}$, and $z=\frac{1\sqrt{2}}{2}+\frac{1\,I\sqrt{2}}{2}$, within the region of integration. To find the residuals for those two poles, use:

```
> z1:=-1/2*sqrt(2)+1/2*I*sqrt(2); z2:=1/2*sqrt(2)+1/2*I*sqrt(2);
```

$$z1 := -\frac{1}{2}\sqrt{2}+\frac{1}{2}I\sqrt{2}$$

$$z2 := \frac{1}{2}\sqrt{2}+\frac{1}{2}I\sqrt{2}$$

```
> ### WARNING: persistent store makes one-argument readlib obsolete
readlib(residue):a[1,-1]:=evalc(residue(F(z),z=z1));a[2,-1]:=evalc(residue(F(z),z=z2));
```

$$a_{1,-1} := \frac{1}{8}\sqrt{2}-\frac{1}{8}I\sqrt{2}$$

$$a_{2,-1} := -\frac{1}{8}\sqrt{2}-\frac{1}{8}I\sqrt{2}$$

The integral of the function $\mathrm{F}(z)=\frac{1}{1+z^4}$ along the closed curve C is obtained using the residual theorem:

```
> integralC:=2*Pi*I*(a[1,-1]+a[2,-1]);
```

$$integralC := \frac{1}{2}\pi\sqrt{2}$$

Notice that the result is independent of R, therefore,

$$\lim_{R\to\infty}\int_C f(z)dz=\frac{\pi\sqrt{2}}{2}.$$

Also, because of the symmetric nature of the function, $\int_0^R \mathrm{f}(x)\,dx=\frac{1}{2}\int_{-R}^R \mathrm{f}(x)\,dx$, thus

$$\int_0^\infty f(x)dx=\frac{1}{2}\lim_{R\to\infty}\int_{-R}^R f(x)dx=\frac{1}{2}\lim_{R\to\infty}\oint_C f(z)dz.$$

i.e.,

```
> integral:=(1/2)*integralC;
```

$$integral := \frac{1}{4}\pi\sqrt{2}$$

Note: With Maple capabilities, the use of the residual theorem to calculate definite integrals as shown in the exercise above is mainly an academic exercise. The integral worked in the previous example can be calculated directly by using:

```
> 'int(1/(1+x^4),x=0..infinity)';%;
```

$$\int_0^\infty \frac{1}{1+x^4}\,dx$$

$$\frac{1}{4}\pi\sqrt{2}$$

Conformal mapping

A complex function *w=f(z)* = $\phi(x,y)+i\,\psi(x,y)$ establishes a unique relationship between points in the x-y plane and those in the $\phi-\psi$ plane in a similar fashion as a real function y = $g(x)$ establishes a unique relationship between points in the x-axis and those of the y axis. A complex function *w=f(z)* is also known as a *mapping* because geometric figures in the x-y plane get transformed, or mapped, into different figures in the $\phi-\psi$ plane.

The following is an example of a mapping provided by the function $w=\frac{1}{z}$ on the intersecting circles C_1: $x^2+y^2=4$, and C_2: $(x-1)^2+y^2=4$. The figures are plotted in the x-y plane as follows:

```
> restart:with(plots):
> eqC1:=x^2+y^2=4:eqC2:=(x-1)^2+y^2=4:                    #Defining curves C1 and C2
> plotC1:=implicitplot(eqC1,x=-3..3,y=-3..3,color = blue):     #Plot curve C1
> plotC2:=implicitplot(eqC2,x=-3..3,y=-3..3,color = green):  #Plot curve C2
> ### WARNING: allvalues now returns a list of symbolic values instead of a sequence of lists of numeric values
solve({eqC1,eqC2},{x,y});IP:=allvalues(%);                    #Intersection points
```

$$\{x=\frac{1}{2}, y=\frac{1}{2}\mathrm{RootOf}(\_Z^2-15)\}$$

$$IP := \{y=\frac{1}{2}\sqrt{15}, x=\frac{1}{2}\}, \{y=-\frac{1}{2}\sqrt{15}, x=\frac{1}{2}\}$$

```
> x1:=rhs(IP[1][2]):y1:=rhs(IP[1][1]):x2:=rhs(IP[2][2]):y2:=rhs(IP[2][1]): #Extracting the points
> z1:=x1+I*y1:z2:=x2+I*y2:plotIP:=complexplot([z1,z2],style=point,symbol=circle, color = black): #Plot points
```

```
> Label1:=textplot([x1+0.1,y1+0.15,`z_1`],align={ABOVE,RIGHT}):
> Label2:=textplot([x2+0.1,y2-0.1,`z_2`],align={BELOW,RIGHT}):
> display({plotC1,plotC2,plotIP,Label1,Label2},scaling=constrained);#Display curves and intersection points
```



Next, we use the mapping $\mathrm{w}(z) = \frac{1}{z}$ to plot the mapped curves:

```
> w:=z->1/z;phi:=evalc(Re(subs(z=x+I*y,w(z))));psi:=evalc(Im(subs(z=x+I*y,w(z)))); #phi and psi functions
```

$$w := z \to \frac{1}{z}$$

$$\phi := \frac{x}{x^2+y^2}$$

$$\psi := -\frac{y}{x^2+y^2}$$

```
> eqC1;expand(eqC2);#Equations for C1 and C2 in x-y
```

$$x^2+y^2=4$$

$$x^2-2x+1+y^2=4$$

From the results for $\phi(x, y)$ and $\psi(x, y)$ we can write:

```
> simplify(phi^2+psi^2);
```

$$\frac{1}{x^2+y^2}$$

i.e., $\phi^2+\psi^2 = \frac{1}{x^2+y^2}$. We redefine the values of ϕ and ψ and replace (x,y) with (ϕ, ψ) to obtain equations for curves C_1 and C_2:

```
> phi:='phi':psi:='psi':eqCC1:=phi^2+psi^2=1/4;            #Curve C1
```

$$eqCC1 := \phi^2+\psi^2 = \frac{1}{4}$$

Also, from $\phi = \frac{x}{x^2+y^2} = x\,(\phi^2+\psi^2)$, it follows that $x = \frac{\phi}{\phi^2+\psi^2}$, which allows us to find the mapped equation for C_2:

```
> eqCC2:=1/(phi^2+psi^2)-2*phi/(phi^2+psi^2)+1=4;     #Curve C2
```

$$eqCC2 := \frac{1}{\phi^2+\psi^2} - 2\frac{\phi}{\phi^2+\psi^2} + 1 = 4$$

```
> plotCC1:=implicitplot(eqCC1,phi=-1..1,psi=-1..1,color=blue):     #Plot for C1 in mapped plane
> plotCC2:=implicitplot(eqCC2,phi=-1..1,psi=-1..1,color=green):  #Plot for C2 in mapped plane
> w1:=evalc(w(z1));w2:=evalc(w(z2));IPP:=[w1,w2];              #Mapping the intersection points
```

$$w1 := \frac{1}{8} - \frac{1}{8} I\sqrt{15}$$

$$w2 := \frac{1}{8} + \frac{1}{8} I\sqrt{15}$$

$$IPP := \left[\frac{1}{8} - \frac{1}{8} I\sqrt{15}, \frac{1}{8} + \frac{1}{8} I\sqrt{15}\right]$$

```
> plotIPP:=complexplot(IPP,phi=-1..1,psi=-1..1,style = point, symbol = circle, color = black): #plot points
> phi1:=Re(w1):psi1:=Im(w1):phi2:=Re(w2):psi2:=Im(w2):
> Label1:=textplot([phi1+0.1,psi1-0.1,`w_1`],align={BELOW,RIGHT}):
> Label2:=textplot([phi2+0.1,psi2+0.1,`w_2`],align={ABOVE,RIGHT}):
> display({plotCC1,plotCC2,plotIPP,Label1,Label2},scaling=constrained);    #Display curves and intersection points
```



The mapping is a *conformal mapping* if the the angle between two arbitrary curves at their point of intersection in the x-y plane is preserved in the curves mapped in the $\phi - \psi$ plane at the corresponding intersection point.   You can verify that is the case for the example worked out above by using the following:

First, the angle of a tangent to C_1at point z_1 is calculated as follows:

```
> eqC1:=x^2+y(x)^2=4;eqC1deriv:=diff(eqC1,x);Dy1:=solve(eqC1deriv,diff(y(x),x));
```

$$eqC1 := x^2 + y(x)^2 = 4$$

$$eqC1deriv := 2\,x + 2\,y(x)\left(\frac{\partial}{\partial x}y(x)\right) = 0$$

$$Dy1 := -\frac{x}{y(x)}$$

```
> m1:=subs({x=x1,y(x)=y1},Dy1);theta1:=evalf((180/Pi)*arctan(m1));   #Angle in degrees
```

$$m1 := -\frac{1}{15}\sqrt{15}$$

$$\theta1 := -14.47751218$$

Next, the angle of a tangent to C_2 at point $z_1$is calculated as follows:

```
> eqC2:=(x-1)^2+y(x)^2=4;eqC2deriv:=diff(eqC2,x);Dy2:=solve(eqC2deriv,diff(y(x),x));
```

$$eqC2 := (x-1)^2 + y(x)^2 = 4$$

$$eqC2deriv := 2\,x - 2 + 2\,y(x)\left(\frac{\partial}{\partial x}\,y(x)\right) = 0$$

$$Dy2 := -\frac{x-1}{y(x)}$$

```
> m2:=subs({x=x1,y(x)=y1},Dy2);theta2:=evalf((180/Pi)*arctan(m2)); #Angle in degrees
```

$$m2 := \frac{1}{15}\sqrt{15}$$

$$\theta2 := 14.47751218$$

Thus, the angle between the two tangents is given by:

```
> Delta[1]:=theta2-theta1;  #Angle in degrees
```

$$\Delta_1 := 28.95502437$$

Now, in the plane $\phi-\psi$, we first find the angle of a tangent at the point of intersection $w_1$as follows:

```
> eqCC1:=phi^2+psi(phi)^2=1/4;eqCC1deriv:=diff(eqCC1,phi);Dpsi1:=solve(eqCC1deriv,diff(psi(phi),phi));
```

$$eqCC1 := \phi^2 + \psi(\phi)^2 = \frac{1}{4}$$

$$eqCC1deriv := 2\,\phi + 2\,\psi(\phi)\left(\frac{\partial}{\partial \phi}\,\psi(\phi)\right) = 0$$

$$Dpsi1 := -\frac{\phi}{\psi(\phi)}$$

```
> m1:=subs({phi=phi1,psi(phi)=psi1},Dpsi1);theta1:=evalf((180/Pi)*arctan(m1));   #Angle in degrees
```

$$m1 := \frac{1}{15}\sqrt{15}$$

$$\theta1 := 14.47751218$$

Next, we find the angle of a tangent at the point of intersection $w_1$as follows:

```
> eqCC2:=1/(phi^2+psi(phi)^2)-2*phi/(phi^2+psi(phi)^2)+1=4;  ;
```

$$eqCC2 := \frac{1}{\phi^2 + \psi(\phi)^2} - 2\,\frac{\phi}{\phi^2 + \psi(\phi)^2} + 1 = 4$$

```
> eqCC2deriv:=diff(eqCC2,phi);Dpsi2:=solve(eqCC2deriv,diff(psi(phi),phi));
```

$$eqCC2deriv := -\frac{2\,\phi + 2\,\psi(\phi)\left(\frac{\partial}{\partial \phi}\,\psi(\phi)\right)}{\left(\phi^2 + \psi(\phi)^2\right)^2} - 2\,\frac{1}{\phi^2 + \psi(\phi)^2} + 2\,\frac{\phi\left(2\,\phi + 2\,\psi(\phi)\left(\frac{\partial}{\partial \phi}\,\psi(\phi)\right)\right)}{\left(\phi^2 + \psi(\phi)^2\right)^2} = 0$$

$$Dpsi2 := -\frac{-\phi+\phi^2-\psi(\phi)^2}{\psi(\phi)\,(-1+2\,\phi)}$$

```
> m2:=subs({phi=phi1,psi(phi)=psi1},Dpsi2);theta2:=evalf((180/Pi)*arctan(m2));  #Angle in degrees
```

$$m2 := \frac{11}{45}\sqrt{15}$$

$$\theta 2 := 43.43253655$$

The angle between the two tangents at the point of intersection is now:

```
> Delta[2]:=theta2-theta1;
```

$$\Delta_2 := 28.95502437$$

To check if the two angles are the same use:

```
> Delta[1]-Delta[2];
```

$$0$$

Thus, we proved that the angles between the two curves C_1 and C_2 are the same in both the x-y and the $\phi-\psi$ planes. The mapping $w=\frac{1}{z}$ is, therefore, conformal.

It can be proven that a mapping defined by a complex function $w=\mathrm{f}(z)$ that is analytical on a region R of the x-y plane is conformal at all points of R.

A catalog of conformal mappings

Some simple conformal mappings are provided by the following functions:

Translation: $w=z+k$, figures in the x-y plane are translated in the direction of the complex number k.

Rotation: $w=\mathbf{e}^{(i\,\alpha)}z$, figures in the x-y plane are rotated by an angle α.

Stretching: $w=k\,z$, figures in the x-y plane are stretched or contracted by the factor k.

Inversion: $w=\frac{1}{z}$, figures in the x-y plane are "inverted" about the unit circle $|z|=1$.

Linear transformation: $w=m\,z+b$, combines stretching with translation.

Bilinear transformation: $w=\frac{a\,z+b}{c\,z+d}$, $a\,d-b\,c\neq 0$, combines translation, rotation, stretching, and inversion.

To illustrate the use of these transformations we will show how two intersecting straight lines in the x-y plane are mapped onto the $\phi-\psi$ plane. The line segments to be used are determined as AB and CD with coordinates:

```
> restart:xA:=-2;yA:=-1;xB:=4;yB:=3;xC:=-3;yC:=3;xD:=3;yD:=-2;
```

$$xA := -2$$

$$yA := -1$$

$$xB := 4$$

$$yB := 3$$

$$xC := -3$$

$$yC := 3$$

$$xD := 3$$

$$yD := -2$$

```
> zA:=xA+I*yA;zB:=xB+I*yB;zC:=xC+I*yC;zD:=xD+I*yD;
```

$$zA := -2 - I$$
$$zB := 4 + 3\,I$$
$$zC := -3 + 3\,I$$
$$zD := 3 - 2\,I$$

```
> with(plots):p1p:=complexplot([zA,zB],x=-5..5,y=-5..5,style=point,symbol=circle, color=blue):
> p1L:=complexplot([zA,zB],x=-5..5,y=-5..5,style=line,color=blue):
> p2p:=complexplot([zC,zD],x=-5..5,y=-5..5,style=point,symbol=circle,color = red):
> p2L:=complexplot([zC,zD],x=-5..5,y=-5..5,style=line,color=red):
> display({p1p,p1L,p2p,p2L});
```



First, we try a translation:

```
> k:=-2-I;w:=z->z+k;wA:=w(zA);wB:=w(zB);wC:=w(zC);wD:=w(zD);
```

$$k := -2 - I$$
$$w := z \to z + k$$
$$wA := -4 - 2\,I$$
$$wB := 2 + 2\,I$$
$$wC := -5 + 2\,I$$
$$wD := 1 - 3\,I$$

```
> p1p:=complexplot([wA,wB],x=-5..5,y=-5..5,style=point,symbol=circle, color=blue):
> p1L:=complexplot([wA,wB],x=-5..5,y=-5..5,style=line,color=blue):
> p2p:=complexplot([wC,wD],x=-5..5,y=-5..5,style=point,symbol=circle,color = red):
> p2L:=complexplot([wC,wD],x=-5..5,y=-5..5,style=line,color=red):
> display({p1p,p1L,p2p,p2L},labels=["phi","psi"]);
```



Next, we show a rotation by an angle $\alpha = \frac{\pi}{3}$:

```
> alpha:=Pi/3;w:=z->z*exp(I*alpha);wA:=evalc(w(zA));wB:=evalc(w(zB));wC:=evalc(w(zC));wD:=evalc(w(zD));
```

$$\alpha := \frac{1}{3}\pi$$

$$w := z \to z\,\mathbf{e}^{(I\,\alpha)}$$

$$wA := -1 + \frac{1}{2}\sqrt{3} + I\left(-\sqrt{3} - \frac{1}{2}\right)$$

$$wB := 2 - \frac{3}{2}\sqrt{3} + I\left(2\sqrt{3} + \frac{3}{2}\right)$$

$$wC := -\frac{3}{2} - \frac{3}{2}\sqrt{3} + I\left(-\frac{3}{2}\sqrt{3} + \frac{3}{2}\right)$$

$$wD := \frac{3}{2} + \sqrt{3} + I\left(\frac{3}{2}\sqrt{3} - 1\right)$$

```
> p1p:=complexplot([wA,wB],x=-5..5,y=-5..5,style=point,symbol=circle, color=blue):
> p1L:=complexplot([wA,wB],x=-5..5,y=-5..5,style=line,color=blue):
> p2p:=complexplot([wC,wD],x=-5..5,y=-5..5,style=point,symbol=circle,color = red):
> p2L:=complexplot([wC,wD],x=-5..5,y=-5..5,style=line,color=red):
>
> display({p1p,p1L,p2p,p2L},labels=["phi","psi"]);
```



The following mapping represents a contraction by a factor $k = .5$:

```
> k:=0.5;w:=z->k*z;wA:=w(zA);wB:=w(zB);wC:=w(zC);wD:=w(zD);
```

$$k := .5$$

$$w := z \to k\, z$$

$$wA := -1.0 - .5\, I$$

$$wB := 2.0 + 1.5\, I$$

$$wC := -1.5 + 1.5\, I$$

$$wD := 1.5 - 1.0\, I$$

```
> p1p:=complexplot([wA,wB],x=-5..5,y=-5..5,style=point,symbol=circle, color=blue):
> p1L:=complexplot([wA,wB],x=-5..5,y=-5..5,style=line,color=blue):
> p2p:=complexplot([wC,wD],x=-5..5,y=-5..5,style=point,symbol=circle,color = red):
> p2L:=complexplot([wC,wD],x=-5..5,y=-5..5,style=line,color=red):
>
> display({p1p,p1L,p2p,p2L},labels=["phi","psi"]);
```



The inversion by $w = \frac{1}{z}$ is shown next:

```
> w:=z->1/z;wA:=evalc(w(zA));wB:=evalc(w(zB));wC:=evalc(w(zC));wD:=evalc(w(zD));
```

$$w := z \to \frac{1}{z}$$

$$wA := -\frac{2}{5} + \frac{1}{5} I$$

$$wB := \frac{4}{25} - \frac{3}{25} I$$

$$wC := -\frac{1}{6} - \frac{1}{6} I$$

$$wD := \frac{3}{13} + \frac{2}{13} I$$

```
> p1p:=complexplot([wA,wB],x=-5..5,y=-5..5,style=point,symbol=circle, color=blue):
> p1L:=complexplot([wA,wB],x=-5..5,y=-5..5,style=line,color=blue):
> p2p:=complexplot([wC,wD],x=-5..5,y=-5..5,style=point,symbol=circle,color = red):
> p2L:=complexplot([wC,wD],x=-5..5,y=-5..5,style=line,color=red):
> display({p1p,p1L,p2p,p2L},labels=["phi","psi"]);
```



A linear transformation is tried next:

```
> w:=z->0.8*z+(1-I);wA:=evalc(w(zA));wB:=evalc(w(zB));wC:=evalc(w(zC));wD:=evalc(w(zD));
```

$$w := z \to .8\, z + 1 - I$$

$$wA := -.6 - 1.8\, I$$

$$wB := 4.2 + 1.4\, I$$

$$wC := -1.4 + 1.4\, I$$

$$wD := 3.4 - 2.6\, I$$

```
> p1p:=complexplot([wA,wB],x=-5..5,y=-5..5,style=point,symbol=circle, color=blue):
> p1L:=complexplot([wA,wB],x=-5..5,y=-5..5,style=line,color=blue):
> p2p:=complexplot([wC,wD],x=-5..5,y=-5..5,style=point,symbol=circle,color = red):
> p2L:=complexplot([wC,wD],x=-5..5,y=-5..5,style=line,color=red):
> display({p1p,p1L,p2p,p2L},labels=["phi","psi"]);
```



A bilinear transformation is provided by:

```
> w:=z->z/(z+2);wA:=evalc(w(zA));wB:=evalc(w(zB));wC:=evalc(w(zC));wD:=evalc(w(zD));
```

$$w := z \to \frac{z}{z+2}$$

$$wA := 1 - 2\,I$$

$$wB := \frac{11}{15} + \frac{2}{15}\,I$$

$$wC := \frac{6}{5} + \frac{3}{5}\,I$$

$$wD := \frac{19}{29} - \frac{4}{29}\,I$$

```
> p1p:=complexplot([wA,wB],x=-5..5,y=-5..5,style=point,symbol=circle, color=blue):
> p1L:=complexplot([wA,wB],x=-5..5,y=-5..5,style=line,color=blue):
> p2p:=complexplot([wC,wD],x=-5..5,y=-5..5,style=point,symbol=circle,color = red):
> p2L:=complexplot([wC,wD],x=-5..5,y=-5..5,style=line,color=red):
> display({p1p,p1L,p2p,p2L},labels=["phi","psi"]);
```



As an additional example we try the transformation $w = z^2$:

```
> w:=z->z^2;wA:=evalc(w(zA));wB:=evalc(w(zB));wC:=evalc(w(zC));wD:=evalc(w(zD));
```

$$w := z \rightarrow z^2$$

$$wA := 3 + 4\,I$$

$$wB := 7 + 24\,I$$

$$wC := -18\,I$$

$$wD := 5 - 12\,I$$

```
> p1p:=complexplot([wA,wB],x=-30..30,y=-30..30,style=point,symbol=circle, color=blue):
```

```
> p1L:=complexplot([wA,wB],x=-5..5,y=-5..5,style=line,color=blue):
```

```
> p2p:=complexplot([wC,wD],x=-5..5,y=-5..5,style=point,symbol=circle,color = red):
```

```
> p2L:=complexplot([wC,wD],x=-5..5,y=-5..5,style=line,color=red):
```

```
> display({p1p,p1L,p2p,p2L},labels=["phi","psi"]);
```



Conformal mappings using the Maple function *conformal*

Maple provides the function *conformal* in the *plots* package to show the conformal mapping $w = \mathrm{f}(z)$ in the $\phi - \psi$ plane corresponding to the rectangle limited by the complex numbers z_1 and z_2. For example, the mapping of the $x - y$ plane in the range -10<x<10, -10<y<10, through the function $\mathrm{f}(z) = \frac{1}{z}$ is given by:

```
> restart:with(plots):conformal(1/z,z=-10-10*I..10+10*I);
```



To determine the range of the mapping you can use a call to the function *conformal* with the rectangular range in the $x-y$ as well as in the $\phi-\psi$ planes. The previous plot, with a restricted range, is shown below:

```
> conformal(1/z,z=-10-10*I..10+10*I,-0.6-0.6*I..0.6+0.6*I);
```



The option *grid* can be used to include more detail in the mapping:

```
> conformal(1/z,z=-10-10*I..10+10*I,-0.6-0.6*I..0.6+0.6*I,grid=[20,20]);
```



The function *conformal* provides a mapping of the lines $x = constant$ and $y = constant$ into the corresponding curves in the $\phi - \psi$ plane. To plot other curves in the $x - y$ plane we need to use the appropriate functions $\phi(x, y)$ and $\psi(x, y)$ and transform the curve accordingly. For example, the curve $y = \sqrt{1 - x^2}$, transformed by using the function $f(z) = \frac{1}{z}$ can be plotted as follows. First, we define the transformation and obtain the corresponding coordinate transformations:

```
> restart:f:=z->1/z;phi:=evalc(Re(subs(z=x+I*y,f(z))));psi:=evalc(Im(subs(z=x+I*y,f(z))));
```

$$f := z \to \frac{1}{z}$$

$$\phi := \frac{x}{x^2 + y^2}$$

$$\psi := -\frac{y}{x^2 + y^2}$$

A plot of that line is produced by using:

```
> with(plots):subs(y=sqrt(1-x^2),[phi,psi,x=-1..1]);plot(%,scaling=constrained);p1:=%:
```

$$[x, -\sqrt{1 - x^2}, x = -1 .. 1]$$



In this plot we have produced a set of (x,y) points in a parametric fashion, with the parameter being x itself. The parametric plot is produced using the function *plot*. Next, we plot the conformal mapping using the function *conformal*:

```
> conformal(1/z,z=-10-10*I..10+10*I,-1-I..1+I,scaling=constrained);p2:=%:
```



Finally, we display the two figures together:

```
> display(p1,p2);
```



The Schwarz-Christoffel transformation

The Schwarz-Christoffel transformation is used to transform a polygon in the $\phi - \psi$ plane into the upper half of the $x - y$ plane. Let the vertices of the polygon in the $\phi - \psi$ plane given by the complex numbers $w_1, w_2, \ldots, w_n$, with the corresponding interior angles given by $\alpha_1, \alpha_2, \ldots, \alpha_n$. The transformation is chosen so that the numbers $w_1, w_2, \ldots, w_n$, are transformed into points $x_1, x_2, \ldots, x_n$ on the real axis of the x-y plane. The Schwarz-Christoffel transformation is defined by

$$\frac{dw}{dz} = K\,(z - x_1)^{\left(\frac{\alpha_1}{\pi} - 1\right)}(z - x_2)^{\left(\frac{\alpha_2}{\pi} - 1\right)} \ldots (z - x_n)^{\left(\frac{\alpha_n}{\pi} - 1\right)}$$

or,

$$w = K \int \prod_{k=1}^{n} (z - x_k)^{\left(\frac{\alpha_k}{\pi} - 1\right)} dz + B,$$

where K and B are complex constants. When using the Schwarz-Christoffel transformation notice that any three of the points $x_1, x_2, \ldots, x_n$, can be chosen arbitrarily. If the last point, x_n, of the points is chosen to be at infinity then the corresponding term drops out of the expression for the transformation.

For example, consider the following polygon in the $\phi - \psi$ plane:

```
> restart:with(plots):phiV:=[-2, -1,0,1,-1];psiV:=[2, 0.5, 0,2,2.5];wList:=seq(phiV[k]+I*psiV[k],k=1..5);
```

$$phiV := [-2, -1, 0, 1, -1]$$

$$psiV := [2, .5, 0, 2, 2.5]$$

$$wList := -2 + 2\,I, -1. + .5\,I, 0, 1 + 2\,I, -1. + 2.5\,I$$

The vertices of the polygon are given by:

```
> wV:=[wList,phiV[1]+I*psiV[1]];
```

$$wV := [-2 + 2\,I, -1. + .5\,I, 0, 1 + 2\,I, -1. + 2.5\,I, -2 + 2\,I]$$

The following commands produce a plot of the polygon in the $\phi - \psi$ plane:

```
> p1:=complexplot(wV,x=-3..2,y=-1..3,style=point,symbol=circle,color = blue):
> p2:=complexplot(wV,x=-3..2,y=-1..3,style=line,color = blue):
> pp1:=textplot([phiV[1]-0.2,psiV[1],`P1`]):pp2:=textplot([phiV[2]-0.3,psiV[2],`P2`]):
> pp3:=textplot([phiV[3]+0.2,psiV[3]-0.2,`P3`]):pp4:=textplot([phiV[4]+0.3,psiV[4],`P4`]):
> pp5:=textplot([phiV[5]-0.2,psiV[5]+0.2,`P5`]):
> plot1:=display({p1,p2,pp1,pp2,pp3,pp4,pp5}):plot1;
```



To find the values of the angles we can use the fact that the angle between two vectors **A** and **B** is given by

$$\cos\theta = \frac{\mathbf{A} \bullet \mathbf{B}}{|\mathbf{A}| \cdot |\mathbf{B}|}.$$

Vectors based on the vertices of the polygon are calculated below and then used to calculate the angles, α_1,α_2, … ,α_n:

```
> with(linalg):
Warning, new definition for norm
Warning, new definition for trace
> P1P5:=[phiV[5]-phiV[1],psiV[5]-psiV[1]];P1P2:=[phiV[2]-phiV[1],psiV[2]-psiV[1]];
```

$$P1P5 := [1, .5]$$

$$P1P2 := [1, -1.5]$$

```
> alpha1:=arccos(dotprod(P1P5,P1P2)/(norm(P1P5,2)*norm(P1P2,2)));
```

$$\alpha 1 := 1.446441332$$

```
> P2P1:=-P1P2;P2P3:=[phiV[3]-phiV[2],psiV[3]-psiV[2]];
```

$$P2P1 := [-1, 1.5]$$

$$P2P3 := [1, -.5]$$

```
> alpha2:=arccos(dotprod(P2P1,P2P3)/(norm(P2P1,2)*norm(P2P3,2)));
```

$$\alpha 2 := 2.622446539$$

```
> P3P2:=-P2P3;P3P4:=[phiV[4]-phiV[3],psiV[4]-psiV[3]];
```

$$P3P2 := [-1, .5]$$

$$P3P4 := [1, 2]$$

```
> alpha3:=arccos(dotprod(P3P2,P3P4)/(norm(P3P2,2)*norm(P3P4,2)));
```

$$\alpha 3 := \frac{1}{2}\pi$$

```
> P4P3:=-P3P4;P4P5:=[phiV[5]-phiV[4],psiV[5]-psiV[4]];
```

$$P4P3 := [-1, -2]$$

$$P4P5 := [-2, .5]$$

```
> alpha4:=arccos(dotprod(P4P3,P4P5)/(norm(P4P3,2)*norm(P4P5,2)));evalf(%);
```

$$\alpha 4 := \arccos(.09701425001\sqrt{5})$$

$$1.352127381$$

```
> P5P4:=-P4P5;P5P1:=-P1P5;
```

$$P5P4 := [2, -.5]$$

$$P5P1 := [-1, -.5]$$

```
> alpha5:=arccos(dotprod(P5P4,P5P1)/(norm(P5P4,2)*norm(P5P1,2)));
```

$$\alpha 5 := 2.432966381$$

The angles α_1, α_2, … , α_n (in radians) should satisfy the condition that $\sum_{k=1}^{n} \alpha_k = (n-2)\pi$. This checks out ok for the angles calculated above:

```
> alpha1+alpha2+alpha3+alpha4+alpha5;evalf(%);evalf(Pi*(5-2));
```

$$6.501854252 + \frac{1}{2}\pi + \arccos(.09701425001\sqrt{5})$$

$$9.424777960$$

$$9.424777962$$

Next, we need to define the values of $x_1, x_2, \ldots, x_n$ in the x-y plane that correspond to the vertices $w_1, w_2, \ldots, w_n$ of the polygon in the $\phi - \psi$ plane. It is recommended that we set one of the vertices at infinity, i.e., let ?, in which case the term

$$(z - x_n)^{\left(\frac{\alpha_n}{\pi} - 1\right)}$$

becomes equal to 1 and essentially drops out of the expression for the Schwarz-Christoffer transformation. The reason for this result is based on introducing the value

$$K = \frac{A}{(-x_n)^{\left(\frac{\alpha_n}{\pi} - 1\right)}}$$

in the expression for the Schwarz-Christoffel transformation. With the value of A as defined, the last term in the expression for the transformation results in

$$\left(\frac{x_n - z}{x_n}\right)^{\left(\frac{\alpha_n}{\pi} - 1\right)} = \left(1 - \frac{z}{x_n}\right)^{\left(\frac{\alpha_n}{\pi} - 1\right)}$$

which becomes 1 as $x_n \to \infty$.

For the case under consideration we have $n = 5$, and we take x_n= ?. Next, we take three arbitrary values of x say, $x_2 = -1$, $x_3 = 0$, and $x_4 = 1$.

```
> xV:=[x1,x2,0,1];alpha:=evalf([alpha1,alpha2,alpha3,alpha4]);
```

$$xV := [x1, x2, 0, 1]$$

$$\alpha := [1.446441332, 2.622446539, 1.570796327, 1.352127381]$$

Thus, the right-hanside of the relationship defining the Schwarz-Christoffel transformation, i.e., the right-hand side of

$$\frac{dw}{dz} = K(z - x_1)^{\left(\frac{\alpha_1}{\pi} - 1\right)}(z - x_2)^{\left(\frac{\alpha_2}{\pi} - 1\right)} \ldots (z - x_n)^{\left(\frac{\alpha_n}{\pi} - 1\right)}$$

for this case will be:

```
> n:=5;'A*product((z-xV[k])^(alpha[k]/Pi-1),k = 1 .. n-1)';%;evalf(%);
```

$$n := 5$$

$$A\left(\prod_{k=1}^{n-1}(z-xV_k)^{\left(\frac{\alpha_k}{\pi}-1\right)}\right)$$

$$A\,(z-x1)^{\left(1.446441332\,\frac{1}{\pi}-1\right)}\,(z-x2)^{\left(2.622446539\,\frac{1}{\pi}-1\right)}\,z^{\left(1.570796327\,\frac{1}{\pi}-1\right)}\,(z-1)^{\left(1.352127381\,\frac{1}{\pi}-1\right)}$$

$$\frac{A}{(z-1.\,x1)^{.5395834244}\,(z-1.\,x2)^{.1652493409}\,z^{.5000000001}\,(z-1.)^{.5696044874}}$$

The integral of this expression does not produce a closed-form expression because of the non-integer powers involved:

```
> 'A*int(product((z-xV[k])^(alpha[k]/Pi-1),k = 1 .. n-1),z)+B';w=%;
```

$$A\int\prod_{k=1}^{n-1}(z-xV_k)^{\left(\frac{\alpha_k}{\pi}-1\right)}\,dz+B$$

$$w=A\int(z-x1)^{\left(1.446441332\,\frac{1}{\pi}-1\right)}\,(z-x2)^{\left(2.622446539\,\frac{1}{\pi}-1\right)}\,z^{\left(1.570796327\,\frac{1}{\pi}-1\right)}\,(z-1)^{\left(1.352127381\,\frac{1}{\pi}-1\right)}\,dz+B$$

The expression obtained above for the transformation $w=\mathrm{f}(z)$ involves four unknowns x_1, x_2, A and B. In theory, we can obtain four equations by replacing the values $w_1=\mathrm{f}(z_1)$, $w_2=\mathrm{f}(z_2)$, $w_3=\mathrm{f}(z_3)$, and $w_4=\mathrm{f}(z_4)$, which can then be solved for the unknowns x_1, x_2, A and B. In practice, the lack of a closed-form expression for the integral above prevents us from obtaining such solution except in some cases as the example that follows.

Schwartz-Christoffel transformation in a semi-infinite domain

Suppose that we want to determine the Schwarz-Christoffel transformation that will transform the upper half of the $x-y$ plane into the semi-infinite domain shown in the left-hand side of the figure below:



The semi-infinite domain MNOP can be thought of a triangle with two vertices at $\phi=-1$ (point N) and $\phi=1$ (point O) and a third vertex at infinity, i.e., points M and P are both located at infinity. The angles at N and O are $\theta_1=\theta_2=\frac{\pi}{2}$. When setting up the Schwarz-Christoffel transformation, we select $x_1=1$ (point N'), $x_2=1$

(point O'), and ?. The following Maple commands are used to set up the transformation:

```
> restart:x:=[-1,1];alpha:=[Pi/2,Pi/2];n:=3;
```

$$x := [-1, 1]$$

$$\alpha := \left[\frac{1}{2}\pi, \frac{1}{2}\pi\right]$$

$$n := 3$$

The right-hand side of the transformation, dw/dz, for this case is:

```
> 'A*product((z-x[k])^(alpha[k]/Pi-1),k=1..n-1)';%;
```

$$A\left(\prod_{k=1}^{n-1}(z-x_k)^{\left(\frac{\alpha_k}{\pi}-1\right)}\right)$$

$$\frac{A}{\sqrt{z+1}\,\sqrt{z-1}}$$

```
> A_*product((z-x[k])^(alpha[k]/Pi-1),k = 1 .. n-1);
```

$$\frac{A\_}{\sqrt{z+1}\,\sqrt{z-1}}$$

The constant A in the previous expression can be replaced by an imaginary constant, say, $A = i\,C$, where C is a real constant, so that we can write the expression as:

$$\frac{A}{\sqrt{z^2-1}} = \frac{i\,C}{\sqrt{z^2-1}} = \frac{C}{\sqrt{1-z^2}}\,.$$

The transformation function $w = \mathrm{f}(z)$ is obtained by integrating the previous expression:

```
> SCTrans:=w=C*int(1/sqrt(1-z^2),z)+B;
```

$$SCTrans := w = C\arcsin(z) + B$$

This result involves two integration constants, A and B, which can be found by using:

```
> Eq1:=subs({z=-1,w=-a},SCTrans);Eq2:=subs({z=1,w=a},SCTrans);
```

$$Eq1 := -a = C\arcsin(-1) + B$$

$$Eq2 := a = C\arcsin(1) + B$$

The solution to this system is:

```
> Sol:=solve({Eq1,Eq2},{B,C});
```

$$Sol := \left\{C = 2\frac{a}{\pi}, B = 0\right\}$$

With these results the expression for the tranformation is:

```
> SCTrans:=subs(Sol,SCTrans);
```

$$SCTrans := w = 2\frac{a\arcsin(z)}{\pi}$$

Which can also be written as:

```
> z=solve(SCTrans,z);
```

$$z = \sin\left(\frac{1}{2}\frac{w\,\pi}{a}\right)$$

To produce a plot of the x-axis onto the $\phi - \psi$ plane we first identify the functions $\phi(x, y)$ and $\psi(x, y)$ from $w = \mathrm{f}(z) = \phi(x, y) + i\,\psi(x, y)$:

```
> x:='x';w:=z->2*a*arcsin(z)/Pi;phi:=evalc(subs(z=x+I*y,Re(w(z))));psi:=evalc(subs(z=x+I*y,Im(w(z))));
```

$$x := x$$

$$w := z \to 2\,\frac{a\,\arcsin(z)}{\pi}$$

$$\phi := 2\,\frac{a\,\arcsin\left(\frac{1}{2}\sqrt{x^2+2\,x+1+y^2}-\frac{1}{2}\sqrt{x^2-2\,x+1+y^2}\right)}{\pi}$$

$$\psi := 2\,a\,\mathrm{csgn}(-I\,x+y)$$
$$\ln\left(\frac{1}{2}\sqrt{x^2+2\,x+1+y^2}+\frac{1}{2}\sqrt{x^2-2\,x+1+y^2}+\sqrt{\left(\frac{1}{2}\sqrt{x^2+2\,x+1+y^2}+\frac{1}{2}\sqrt{x^2-2\,x+1+y^2}\right)^2-1}\right)/\pi$$

Notice in the result for ψ includes a function called *csgn* (complex *sign*) which, when applied to a complex number z, determines the location of the number in the left-side or right-side of the plane. (To find more about this function use:

```
> ?csgn
```

An attempt to plot the transformation of the x-axis (y = 0) using the functions ϕ and ψ follows:

```
> with(plots):a:=1;subs(y=0,[phi,psi,x=-10..10]);plot(%,scaling=constrained);p1:=%:
```

$$a := 1$$

$$\left[2\,\frac{\arcsin\left(\frac{1}{2}\sqrt{x^2+2\,x+1}-\frac{1}{2}\sqrt{x^2-2\,x+1}\right)}{\pi},\right.$$
$$\left.2\,\frac{\mathrm{csgn}(-I\,x)\,\ln\left(\frac{1}{2}\sqrt{x^2+2\,x+1}+\frac{1}{2}\sqrt{x^2-2\,x+1}+\sqrt{\left(\frac{1}{2}\sqrt{x^2+2\,x+1}+\frac{1}{2}\sqrt{x^2-2\,x+1}\right)^2-1}\right)}{\pi}, x = -10\,..\,10\right]$$



Eliminating the function *csgn* in the definition of ψ produces the desired result:

```
> psi:=2*ln(1/2*sqrt(x^2+2*x+1+y^2)+1/2*sqrt(x^2-2*x+1+y^2)+sqrt((1/2*sqrt(x^2+2*x+1+y^2)+1/2*sqrt(
```

```
x^2-2*x+1+y^2))^2-1))/Pi;
```

$$\psi := 2\,\frac{\ln\left(\frac{1}{2}\sqrt{x^2+2\,x+1+y^2}+\frac{1}{2}\sqrt{x^2-2\,x+1+y^2}+\sqrt{\left(\frac{1}{2}\sqrt{x^2+2\,x+1+y^2}+\frac{1}{2}\sqrt{x^2-2\,x+1+y^2}\right)^2-1}\right)}{\pi}$$

```
> with(plots):subs(y=0,[phi,psi,x=-10..10]):plot(%,scaling=constrained);p1:=%:
```



The Jukowsky transformation and applications in potential fluid flow

The Jukowsky transfomation is given, in general, by the function

$$\mathrm{w}(z) = \mathbf{e}^{(-i\,\theta)}\left(z+\frac{a^2}{z}\right),$$

where a and θ are constant. The Jukowsky transformation is typically used in potential fluid flow applications because it allows the mapping of a circle or cylinder in the $x-y$ plane into a number of airfoil shapes (i.e., wing cross-sections). The complex potential of the flow about the cylinder in the $x-y$ can be transformed by the Jukowsky mapping and used to produce information on the streamlines and velocity vectors of the corresponding flow about an airfoil.

The following example shows the application of the Jukowsky mapping on a circle of radius R = 2, with a = 1 and θ = 0.

```
> restart:w:=z->exp(-I*theta)*(z
  +a^2/z);phi:=evalc(subs(z=x+I*y,Re(w(z))));psi:=evalc(subs(z=x+I*y,Im(w(z))));
```

$$w := z \rightarrow \mathbf{e}^{(-I\,\theta)}\left(z+\frac{a^2}{z}\right)$$

$$\phi := \cos(\theta)\left(x+\frac{a^2\,x}{x^2+y^2}\right)+\sin(\theta)\left(y-\frac{a^2\,y}{x^2+y^2}\right)$$

$$\psi := -\sin(\theta)\left(x+\frac{a^2\,x}{x^2+y^2}\right)+\cos(\theta)\left(y-\frac{a^2\,y}{x^2+y^2}\right)$$

Let's transform the equations describing the circle of radius R and centered at (x_0, y_0), i.e.,

$y_1(x) = y_0 + \sqrt{R^2 - (x - x0)^2}$, and $y_2(x) = y_0 - \sqrt{R^2 - (x - x_0)^2}$, through the mapping indicated above.

```
> theta:=0;a:=1;R:=2;x0:=0;y0:=1;
```

$$\theta := 0$$
$$a := 1$$
$$R := 2$$
$$x0 := 0$$
$$y0 := 1$$

```
> with(plots):p1:=plot(subs(y=y0+sqrt(R^2-(x-x0)^2),[phi,psi,x=x0-R..x0+R]),scaling=constrained):
```

```
> p2:=plot(subs(y=y0-sqrt(R^2-(x-x0)^2),[phi,psi,x=x0-R..x0+R]),scaling=constrained):
```

```
> display({p1,p2},axes=boxed);
```



Additional examples are shown in the figure below:



θ = 0, a = 1, R = 2, x0 = 1, y0 = 0



$\theta = \pi/10$, a = 2, R = 4, x0 = 1, y0 = 0



$\theta = \pi/10$, a = 2, R = 3, x0 = 1, y0 = 0



$\theta = \pi/10$, a = 1, R = 1, x0 = 0.05, y0 = 0.5

Exercises

In problems [1] through [6], use the following definitions:

$$z_1 = -3 + 2i,\ z_2 = 5 - i,\ z_3 = -4+3i,\ z_4 = -2-4i$$

[1]. Determine the following magnitudes and arguments:
(a) $|z_1|$ (b) $\mathrm{Arg}(z_1)$ (c) $|z_1|$ (d) $\mathrm{Arg}(z_1)$
(e) $|z_3|$ (f) $\mathrm{Arg}(z_3)$ (g) $|z_4|$ (h) $\mathrm{Arg}(z_4)$

[2]. Write the following complex numbers in polar form:
(a) $3-5i$ (b) $4-4i$ (c) $3-5i$ (d) $-2-6i$
(e) $-5+6i$ (f) $-\pi+3i$ (g) $(5-i)/2$ (h) $7+5i$

[3]. Plot the complex numbers of problem [2] in the x-y plane using the function *complexplot*.

[4]. Determine the result of the following complex number operations:
(a) $z_1+3\cdot z_2$ (b) $z_1-z_2+4\cdot z_3$ (c) $(z_1-2)\cdot(z_4-z_3)$ (d) $z_4/z_2 + z_3/z_1$
(e) $z_1\cdot z_2\cdot z_3$ (f) $(z_1-2z_2)\cdot(z_3/\pi)$ (g) $z_1\cdot z_2 -1/z_3$ (h) $(z_2+z_3)/(z_1+3\cdot z_2)$

[5]. Determine the result of the following complex number operations:
(a) $\overline{z_1}+z_1$ (b) $\overline{z_2}\cdot z_2$ (c) $z_3/\overline{z_3}$ (d) $(\overline{z_1}+\overline{z_2})/(z_3-5\overline{z_4})$
(e) $\overline{z_1}-z_1$ (f) $2(\overline{z_1}-z_1)$ (g) $1/\overline{z_2}+1/z_2$ (h) $|z_1|(\overline{z_2}+z_3)$

[6]. Determine the result of the following complex number operations:
(a) z_3^3 (b) $z_2(z_1-2z_3)^2$ (c) z_2/z_3^4 (d) z_3^2/z_1^3
(e) $z_1+1/z_2+1/z_3^2$ (f) $(1+1/z_2)^3$ (g) $z_1^3+(z_2-z_3)^2$ (h) $(z_2+1)^2/z_3$

[7]. Solve for z in the following equations:
(a) $z^2+3-2i = 0$ (b) $(z+1)^2 = 3^{1/2}$ (c) $z^3-i = 0$ (d) $z^4+(i-2)^3 = 0$
(e) $z^2+2z=4$ (f) $z^3=-1$ (g) $1/(z+1)^3 = i$ (h) $z(z-1)=2$

[8]. For the complex functions *f(z)* shown below determine the real and imaginary parts, $\Phi(x,y)$ and $\Psi(x,y)$, and plot the surfaces corresponding to those functions using *plot3d*:
(a) $f(z) = z+1/z$ (b) $f(z) = \exp(z/\pi)$ (c) $f(z) = \ln(z-2i)$ (d) $f(z) = (1+z)^2$
(e) $f(z) = z/(z^2+1)$ (f) $f(z) = z^{1/2}$ (g) $f(z) = \sin(z)$ (h) $f(z) = \cosh(z)$

[9]. After determining the real and imaginary parts for the functions of problem [8], check if the functions are analytic by testing the Cauchy-Riemann conditions:

$$\frac{\partial \Phi}{\partial x} = \frac{\partial \Psi}{\partial y}, \qquad \frac{\partial \Phi}{\partial y} = -\frac{\partial \Psi}{\partial x}.$$

[10]. The functions $u=u(x,y)$ given below represent the x-velocity component of a potential flow in the x-y plane. Using the continuity equation determine an expression for the y-velocity component, $v=v(x,y)$. Also, determine the velocity potential, $\phi(x,y)$, and stream function, $\psi(x,y)$, for the flow:
(a) $u(x,y) = -2xy/(x^2+y^2)^2$ (b) $u(x,y) = x/(x^2+y^2)$
(c) $u(x,y) = U$ (a constant) (d) $u(x,y) = 0$

[11]. If the functions of problem [8] represent the complex potential of a two-dimensional potential flow, determine the complex velocity, *w(z)*, and the x- and y-velocity components of the flow, u and v.
[12]. Plot contours of the velocity potential and the stream function of the complex potentials f(z) of problem [8]. These plots are known as the flow net of the flow.

[13]. Find the velocity potential and stream function for the combination of two vortices rotating in opposite directions, both of strength G = 1. The clockwise vortex is located at z = +i, while the counterclockwise vortex is locates at z = -i. Plot the flow net for this flow.

[14]. Find the velocity potential for the flow that results from combining the flow of problem [13] with a uniform flow u=U in the positive x-direction. Plot the flow net for this flow.

[15]. Plot the vector field for the velocities described by the complex potentials *f(z)* of problem [8].

[16]. Expand the functions of problem [8] using a Taylor series about z = 0.

[17]. Expand the following functions *f(z)* using Laurent series about the points z_0:
(a) $f(z) = \sin(z)/(z^2+2z+2)$, $z_0 = -(1+i)$ (b) $f(z) = e^z/(z-i)^2$, $z_0 = i$
(c) f(z) = z/(z-1)^2, $z_0 = 1$ (d) $f(z) = \cos(z)/(z-\pi)$, $z_0 = \pi$

[18]. Calculate the integral of the function *f(z)* for the limits of integration z_1 and z_2 along the curve indicated:
(a) f(z) = exp(z), $z_1 = 2+2i$, $z_2 = -5-5i$, C: y = x
(b) f(z) = sin(z), $z_1 = 2i$, $z_2 = 1+3i$, C: $y = x^2+2$
(c) $f(z) = z^2$, $z_1 = i$, $z_2 = 5+6i$, C: y = x+1
(d) $f(z) = (1+z)^2$, $z_1 = 2+i$, $z_2 = 4+2i$, C: y = x/2

[19]. Calculate the integral of the function *f(z)* about the closed curve C using Cauchy's integral formulae:
(a) $f(z) = z/(z-1)^2$, C: $|z-1|<5$ (b) f(z) = (1+z)/(z+2i), C: $|z+2i|<5$
(c) $f(z) = \sin(z)/(z-\pi/2^2)$, C: $|z-\pi/2|< 3$ (d) $f(z) = e^z/(a-z)^3$, C: $|z-a|<a$, $a>0$

[20]. Determine the residues for the function *f(z)* if the function is defined within a simple closed curve containing the function's singularities:
(a) $f(z) = z/(z^2+4)$ (b) $f(z) = \sin(z)/(z^4-4z^3-11z^2+66z-72)$
(c) $f(z) = z^2/(z-2)^3(z+1)$ (d) $f(z) = \sin(z+\pi)/(z^4-16)$

[21]. Calculate the integral of the functions *f(z)* from problem [20] on the closed curve C that contains the function's singularities.

[22]. Use complex variables methods to calculate the following definite integrals:
(a) $\int_0^\infty \frac{dx}{x^2-1}$ (b) $\int_{-\infty}^\infty \frac{dx}{x^2+1}$ (c) $\int_{-\infty}^\infty \frac{dx}{(x^2+1)^2}$

[23]. Let C be the upper part of the circle $y = \sqrt{1-x^2}$, *-1<x<1*, in the x-y plane. Plot the mapping of C into the $\phi-\psi$ plane using the following transformations:
(a) $f(z) = z^2$ (b) $f(z) = z^{1/2}$ (c) f(z) = sin(z) (d) f(z) = exp(z)

[24]. Determine the function that maps the region in the $\phi-\psi$ plane into the upper half of the x-y plane:



[25]. Plot the transformation of a circle of radius R, centered at (x_0,y_0), resulting from the Jukowsky transformation with the following parameters:

(a) $\theta = \pi/10$, a = 1, R = 1, x_0 = 1, y_0 = -1 (b) $\theta = \pi/6$, a = 2, R = 5, x_0 = 0.5, y_0 = -0.5

(c) $\theta = \pi/20$, a = 3, R = 2, x_0 = 0, y_0 = -1 (d) $\theta = \pi/3$, a = 4, R = 1, x_0 = 1, y_0 = 0

REFERENCES

Farlow, Stanley J., 1982, "*Partial Differential Equations for Scientists and Engineers*," Dover Publications Inc., New York.

Friedman, B., 1956 (reissued 1990), "*Principles and Techniques of Applied Mathematics*," Dover Publications Inc., New York.

Gullberg, J., 1997, "*Mathematics – From the Birth of Numbers*," W. W. Norton & Company, New York.

Harris, J.W., and H. Stocker, 1998, "*Handbook of Mathematics and Computational Science*," Springer, New York.

Heal, K.M., M.L. Hansen, and K. M. Rickard, 1998, "*Maple V - Learning Guide*," Springer, New York.

Kamerich, E., 1999, "*A Guide to Maple*," Springer, New York.

Keener, J. P., 1998, "*Principles of Applied Mathematics*," Perseus Books, Reading, Massachussetts.

Kottegoda, N. T., and R. Rosso, 1997, "*Probability, Statistics, and Reliability for Civil and Environmental Engineers*," The Mc-Graw Hill Companies, Inc., New York.

Kreysig, E., 1983, "*Advanced Engineering Mathematics – Fifth Edition*," John Wiley & Sons, New York.

Monangan, M.B., K.O. Geddes, K.M. Heal, G. Labahn, and S.M. Vorkoetter, 1998, "*Maple V - Programming Guide*," Springer, New York.

Newland, D.E., 1993, "*An Introduction to Random Vibrations, Spectral & Wavelet Analysis – Third Edition*," Longman Scientific and Technical, New York.

Robertson J.S., 1996, "*Engineering Mathematics with Maple*," Mc Graw Hill, Inc., New York.

Schwartz, D.I., 1999, "*Introduction to Maple*", E-Source, Prentice Hall, Upper Saddle River, New Jersey.

Spiegel, M. R., 1971 (second printing, 1999), "*Schaum's Outline of Theory and Problems of Advanced Mathematics for Engineers and Scientists*," Schaum's Outline Series, McGraw-Hill, New York.

Tinker, M. and R. Lambourne, 2000, "*Further Mathematics for the Physical Sciences*," John Wiley & Sons, LTD., Chichester, U.K.

Tveito, A. and R. Winther, 1998, "*Introduction to Partial Differential Equations - A Computational Approach,*" Texts in Applied Mathematics 29, Springer, New York.

Urroz, G., 2000, "*Science and Engineering Mathematics with the HP 49 G - Volumes I & II*", www.greatunpublished.com, Charleston, S.C.